

The Agda Universal Algebra Library

Part 2: Structure

Homomorphisms, terms, classes of algebras, subalgebras, and homomorphic images

William DeMeo   

Department of Algebra, Charles University in Prague

Abstract

The Agda Universal Algebra Library (UALib) is a library of types and programs (theorems and proofs) we developed to formalize the foundations of universal algebra in dependent type theory using the Agda programming language and proof assistant. The UALib includes a substantial collection of definitions, theorems, and proofs from universal algebra, equational logic, and model theory, and as such provides many examples that exhibit the power of inductive and dependent types for representing and reasoning about mathematical structures and equational theories. In this paper, we describe the types and proofs of the UALib that concern homomorphisms, terms, and subalgebras.

2012 ACM Subject Classification Theory of computation \rightarrow Logic and verification; Computing methodologies \rightarrow Representation of mathematical objects; Theory of computation \rightarrow Type theory

Keywords and phrases Agda, constructive mathematics, dependent types, equational logic, extensionality, formalization of mathematics, model theory, type theory, universal algebra

Related Version hosted on arXiv

Part 1, Part 3: http://arxiv.org/a/demeo_w_1

Supplementary Material

Documentation: ualib.org

Software: <https://gitlab.com/ualib/ualib.gitlab.io.git>

Contents

1	Introduction	2
1.1	Motivation	2
1.2	Attributions and Contributions	2
1.3	Organization of the paper	2
1.4	Resources	3
2	Homomorphism Types	4
2.1	Basic definitions	4
2.2	Homomorphism Theorems	9
2.3	Isomorphisms	15
2.4	Homomorphic Images	19
3	Types for Terms	20
3.1	Basic Definitions	20
3.2	Term Operations	23
4	Subalgebra Types	25
4.1	Subuniverses	25
4.2	Subalgebras	29
5	Concluding Remarks	33



This work and the Agda Universal Algebra Library by William DeMeo is licensed under a Creative Commons Attribution-ShareAlike 4.0 International License.



© 2021 William DeMeo. Based on work at <https://gitlab.com/ualib/ualib.gitlab.io>.
Compiled with **xelatex** on 17 Mar 2021 at 17:13.

1 Introduction

To support formalization in type theory of research level mathematics in universal algebra and related fields, we present the Agda Universal Algebra Library ([AgdaUALib](#)), a software library containing formal statements and proofs of the core definitions and results of universal algebra. The [UALib](#) is written in [Agda](#) [8], a programming language and proof assistant based on Martin-Löf Type Theory that not only supports dependent and inductive types, as well as proof tactics for proving things about the objects that inhabit these types.

1.1 Motivation

The seminal idea for the [AgdaUALib](#) project was the observation that, on the one hand, a number of fundamental constructions in universal algebra can be defined recursively, and theorems about them proved by structural induction, while, on the other hand, inductive and dependent types make possible very precise formal representations of recursively defined objects, which often admit elegant constructive proofs of properties of such objects. An important feature of such proofs in type theory is that they are total functional programs and, as such, they are computable, composable, and machine-verifiable.

Finally, our own research experience has taught us that a proof assistant and programming language (like Agda), when equipped with specialized libraries and domain-specific tactics to automate the proof idioms of our field, can be an extremely powerful and effective asset. As such we believe that proof assistants and their supporting libraries will eventually become indispensable tools in the working mathematician’s toolkit.

1.2 Attributions and Contributions

The mathematical results described in this paper have well known *informal* proofs. Our main contribution is the formalization, mechanization, and verification of the statements and proofs of these results in dependent type theory using Agda.

Unless explicitly stated otherwise, the Agda source code described in this paper is due to the author, with the following caveat: the [UALib](#) depends on the [Type Topology](#) library of [Martín Escardó](#) [5]. Each dependency is carefully accounted for and mentioned in this paper. For the sake of completeness and clarity, and to keep the paper mostly self-contained, we repeat some definitions from the [Type Topology](#) library, but in each instance we cite the original source.¹

1.3 Organization of the paper

In this paper we limit ourselves to the presentation of the middle third of the [UALib](#), which includes types for representing homomorphisms, terms, and subalgebras. This limitation will allow us the space required to discuss some of the more interesting type theoretic and foundational issues that arise when developing a library of this kind and when attempting to represent advanced mathematical notions in type theory and formalize them in Agda.

This is the second installment of a three-part series of papers describing the [AgdaUALib](#). The first part is [3] which covers relations, algebras, congruences, and quotients. The third part will cover free algebras, equational classes of algebras (i.e., varieties), and Birkhoff’s HSP theorem.

¹ In the [UALib](#), such instances occur only inside hidden modules that are never actually used, followed immediately by a statement that imports the code in question from its original source.

The present paper is divided into three main parts (§2, §3, §4). The first of these introduces types representing *homomorphisms* from one algebra to another, and presents a formal statement and proof of the first fundamental theorem about homomorphisms, known as the *First Isomorphism Theorem*, as well as a version of the so-called *Second Isomorphism Theorem*. This is followed by dependent type definitions for representing *isomorphisms* and *homomorphic images* of algebraic structures.

In Section 3 we define inductive types to represent *terms* and the *term algebra* in a given signature. We prove the *universal property* of the term algebra which is the fact that term algebra is *free* (or *initial*) in the class of all algebras in the given signature. We define types that denote the interpretation of a term in an algebra type, called a *term operation*, including the interpretation of terms in *arbitrary products* of algebras (§3.2.1). We conclude § 3 with a subsection on the compatibility of terms with basic operations and congruence relations.

Section 4 presents inductive and dependent types for representing subuniverses and subalgebras of a given algebra. Here we define an inductive type that represents the *subuniverse generated by X*, for a given predicate $X : \text{Pred} \mid \mathbf{A} \mid _$,² and we use this type to formalize a few basic subuniverse lemmas. We also define types that pertain to arbitrary classes of algebras. In particular, in Subsection 4.2 on subalgebras, we define a type that represents the assertion that a given algebra is a subalgebra of some member of a class of algebras.

1.4 Resources

We conclude this introduction with some pointers to helpful reference materials. For the required background in Universal Algebra, we recommend the textbook by Clifford Bergman [1]. For the type theory background, we recommend the HoTT Book [9] and Escardó’s [Introduction to Univalent Foundations of Mathematics with Agda](#) [5].

The following are informed the development of the UALib and are highly recommended.

- [Introduction to Univalent Foundations of Mathematics with Agda](#), Escardó [5]
- [Dependent Types at Work](#), Bove and Dybjer [2]
- [Dependently Typed Programming in Agda](#), Norell and Chapman [7]
- [Formalization of Universal Algebra in Agda](#), Gunther, Gadea, Pagano [6]
- [Programming Languages Foundations in Agda](#) [13]

More information about AgdaUALib can be obtained from the following official sources.

- ualib.org (the web site) documents every line of code in the library.
- gitlab.com/ualib/ualib.gitlab.io (the source code) AgdaUALib is open source.³
- [The Agda UALib, Part 1: equality, extensionality, truncation, and dependent types for relations and algebras](#) [3].
- [The Agda UALib, Part 3: free algebras, equational classes, and Birkhoff’s theorem](#) [4].

The first item links to the official UALib html documentation which includes complete proofs of every theorem we mention here, and much more, including the Agda modules covered in the first and third installments of this series of papers on the UALib.

Finally, readers will get much more out of reading the paper if they download the AgdaUALib from <https://gitlab.com/ualib/ualib.gitlab.io>, install the library, and try it out for themselves.

² As we learned in [3], such X represents a subset of the domain of the algebra \mathbf{A} .

³ License: [Creative Commons Attribution-ShareAlike 4.0 International License](#).

2 Homomorphism Types

In this section we define types that represent some of the most important concepts from general (universal) algebra. Note that the Agda modules we describe here, and in succeeding sections depend on and import the modules that were presented in Part 1 of this series of three papers describing the `AgdaUALib` (see [3]).

We begin in Subsection 2.1 with the basic definition of *homomorphism*. In §2.2 we formalize the statement and proof of the first fundamental theorem about homomorphisms, which is sometimes referred to as the *First Isomorphism Theorem*. This is followed by §2.3, in which we define the type of *isomorphisms* between algebraic structures. Finally, in §2.4, we define types that manifest the notion of *homomorphic image*.

2.1 Basic definitions

This section presents the `Homomorphisms.Basic` module of the `AgdaUALib`, slightly abridged.⁴ Since this is the first module we introduce in the installment of our series of papers documenting the `AgdaUALib`, we will begin by showing the start of the module in full. In later modules, we will leave such details implicit. Here is how the file `Homomorphisms/Basic.lagda` of the `UALib` begins.

```
{-# OPTIONS -without-K -exact-split -safe #-}

open import Algebras.Signatures using (Signature; Ⓞ; ℳ)
open import MGS-Subsingleton-Theorems using (global-dfunext)

module Homomorphisms.Basic {S : Signature Ⓞ ℳ}{gfe : global-dfunext} where

open import Algebras.Congruences{S = S} public
open import MGS-MLTT using (≡; ≡; ▯) public
```

The `OPTIONS` *pragma* sets some parameters that specify the type theoretic foundations that we assume. These are discussed in [3, §2.1], but let’s briefly review: `-without-K` disables *Streicher’s K axiom* (see [10]); `-exact-split` makes Agda accept only definitions that use *definitional* equalities (see [12]); `-safe` ensures that nothing is postulated outright, so that every non-MLTT axiom has to be an explicit assumption (see [11] and [12]).⁵

2.1.1 Homomorphisms

If **A** and **B** are *S*-algebras, then a *homomorphism* is a function $h : | \mathbf{A} | \rightarrow | \mathbf{B} |$ from the domain of **A** to the domain of **B** that is *compatible* (or *commutes*) with all of the basic operations of the signature; that is, for all operation symbols $f : | S |$ and all tuples $a : || S || f \rightarrow | \mathbf{A} |$, the following equality holds:⁶

$$h ((f \wedge \mathbf{A}) a) \equiv (f \wedge \mathbf{B}) (h \circ a).$$

Instead of “homomorphism,” we sometimes use the nickname “hom” to refer to such a map.

⁴ For unabridged docs and source code see <https://ualib.gitlab.io/Homomorphisms.Basic.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Homomorphisms/Basic.lagda>.

⁵ As in [3], MLTT stands for Martin-Löf Type Theory.

⁶ Recall, $h \circ a$ is the tuple whose *i*-th component is $h (a \ i)$.

To formalize the notion of homomorphism in type theory, we first define a type representing the assertion that a function $h : | \mathbf{A} | \rightarrow | \mathbf{B} |$, from the domain of \mathbf{A} to the domain of \mathbf{B} , *commutes* (or is *compatible*) with an operation f , interpreted in the algebras \mathbf{A} and \mathbf{B} . Pleasingly, the defining equation of the previous paragraph can be expressed in Agda unadulterated.⁷

```
module _ { $\mathcal{U} \mathcal{W} : \text{Universe}$ } where

compatible-op-map : ( $\mathbf{A} : \text{Algebra } \mathcal{U} S$ )( $\mathbf{B} : \text{Algebra } \mathcal{W} S$ )
  ( $f : | S |$ )( $h : | \mathbf{A} | \rightarrow | \mathbf{B} |$ )  $\rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W} \cdot$ 
compatible-op-map  $\mathbf{A} \mathbf{B} f h = \forall a \rightarrow h ((f \wedge \mathbf{A}) a) \equiv (f \wedge \mathbf{B}) (h \circ a)$ 
```

Note the appearance of the shorthand $\forall a$ in the definition of `compatible-op-map`. We can get away with this in place of $(a : | S | \parallel f \rightarrow | \mathbf{A} |)$ since Agda is able to infer that the a here must be a tuple on $| \mathbf{A} |$ of “length” $| S | \parallel f$ (the arity of f).

We now define the type `hom $\mathbf{A} \mathbf{B}$` of homomorphisms from \mathbf{A} to \mathbf{B} by first defining the property `is-homomorphism`.

```
is-homomorphism : ( $\mathbf{A} : \text{Algebra } \mathcal{U} S$ )( $\mathbf{B} : \text{Algebra } \mathcal{W} S$ )  $\rightarrow (| \mathbf{A} | \rightarrow | \mathbf{B} |) \rightarrow \mathcal{O} \sqcup \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W} \cdot$ 
is-homomorphism  $\mathbf{A} \mathbf{B} g = \forall f \rightarrow \text{compatible-op-map } \mathbf{A} \mathbf{B} f g$ 

hom :  $\text{Algebra } \mathcal{U} S \rightarrow \text{Algebra } \mathcal{W} S \rightarrow \mathcal{O} \sqcup \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W} \cdot$ 
hom  $\mathbf{A} \mathbf{B} = \Sigma g : (| \mathbf{A} | \rightarrow | \mathbf{B} |) , \text{is-homomorphism } \mathbf{A} \mathbf{B} g$ 
```

2.1.2 Examples

Here we give a few very special examples of homomorphisms. In each case, the function in question commutes with the basic operations of *all* algebras and so, no matter the algebras involved, is always a homomorphism (trivially).

The most obvious example of a homomorphism is the identity map, which is proved to be a homomorphism as follows.

```
id : { $\mathcal{U} : \text{Universe}$ } ( $A : \text{Algebra } \mathcal{U} S$ )  $\rightarrow \text{hom } A A$ 
id _ = ( $\lambda x \rightarrow x$ ) ,  $\lambda \_ \rightarrow \text{refl}$ 

id-is-hom : { $\mathcal{U} : \text{Universe}$ }{ $\mathbf{A} : \text{Algebra } \mathcal{U} S$ }  $\rightarrow \text{is-homomorphism } \mathbf{A} \mathbf{A} (\text{id} | \mathbf{A} |)$ 
id-is-hom =  $\lambda \_ \rightarrow \text{refl}$ 
```

Next, `lift` and `lower`, defined in the `Prelude.Lifts` module (see [3, §2.5]),⁸ are (the maps of) homomorphisms. Again, the proof is trivial in each case.

```
lift-is-hom : { $\mathbf{A} : \text{Algebra } \mathcal{U} S$ }{ $\mathcal{W} : \text{Universe}$ }  $\rightarrow \text{is-homomorphism } \mathbf{A} (\text{lift-alg } \mathbf{A} \mathcal{W}) \text{ lift}$ 
lift-is-hom _ _ =  $\text{refl}$ 

lift : { $\mathbf{A} : \text{Algebra } \mathcal{U} S$ }{ $\mathcal{W} : \text{Universe}$ }  $\rightarrow \text{hom } \mathbf{A} (\text{lift-alg } \mathbf{A} \mathcal{W})$ 
lift = ( $\text{lift} , \text{lift-is-hom}$ )
```

⁷ Here we put the definition inside an *anonymous module*, which starts with the `module` keyword followed by an underscore (instead of a module name). The purpose is simply to move the postulated typing judgments (the “parameters” of the module, e.g., $\mathcal{U} \mathcal{W} : \text{Universe}$) out of the way so they don’t obfuscate the definitions inside the module. In descriptions of the UALib, such as the present paper, we usually don’t show the module declarations unless we wish to emphasize the typing judgments that are postulated in the module declaration.

⁸ or <https://ualib.gitlab.io/Prelude.Lifts.html>.

```
lower-is-hom : {A : Algebra U S}{W : Universe} → is-homomorphism (lift-alg A W) A lower
lower-is-hom _ _ = refl
```

```
lower : (A : Algebra U S){W : Universe} → hom (lift-alg A W) A
lower A = (lower , lower-is-hom{A})
```

2.1.3 Monomorphisms and epimorphisms

We represent *monomorphisms* (injective homomorphisms) and *epimorphisms* (surjective homomorphisms) by the following types.

```
is-monomorphism : (A : Algebra U S)(B : Algebra W S) → (| A | → | B |) → 0 ⊔ V ⊔ U ⊔ W ·
is-monomorphism A B g = is-homomorphism A B g × Monic g
```

```
mon : Algebra U S → Algebra W S → 0 ⊔ V ⊔ U ⊔ W ·
mon A B = Σ g : (| A | → | B |) , is-monomorphism A B g
```

```
is-epimorphism : (A : Algebra U S)(B : Algebra W S) → (| A | → | B |) → 0 ⊔ V ⊔ U ⊔ W ·
is-epimorphism A B g = is-homomorphism A B g × Epic g
```

```
epi : Algebra U S → Algebra W S → 0 ⊔ V ⊔ U ⊔ W ·
epi A B = Σ g : (| A | → | B |) , is-epimorphism A B g
```

It will be convenient to have a function that takes an inhabitant of ‘mon’ (or ‘epi’) and extracts the homomorphism part, or `*hom reduct*`, that is, the pair consisting of the map and a proof that the map is a homomorphism.

```
mon-to-hom : (A : Algebra U S)(B : Algebra W S) → mon A B → hom A B
mon-to-hom A φ = | φ | , fst || φ ||
```

```
epi-to-hom : {A : Algebra U S}(B : Algebra W S) → epi A B → hom A B
epi-to-hom _ φ = | φ | , fst || φ ||
```

2.1.4 Equalizers in Agda

Recall, the equalizer of two functions (resp., homomorphisms) $g h : A \rightarrow B$ is the subset of A on which the values of the functions g and h agree. We define the equalizer of functions and homomorphisms in the `UALib` as follows.

```
E : {B : Algebra W S}(g h : | A | → | B |) → Pred | A | W
E g h x = g x ≡ h x
```

```
Ehom : (B : Algebra W S)(g h : hom A B) → Pred | A | W
Ehom _ g h x = | g | x ≡ | h | x
```

We will define subuniverses in the `Algebras.Subuniverses` module, but we note here that the equalizer of homomorphisms from \mathbf{A} to \mathbf{B} will turn out to be subuniverse of \mathbf{A} . Indeed, this is easily proved as follows.

```
Ehom-closed : (B : Algebra W S)(g h : hom A B)
→      ∀ f a → (∀ x → a x ∈ Ehom B g h)
-----
→      | g | ((f ^ A) a) ≡ | h | ((f ^ A) a)
```

$$\begin{aligned}
\text{Ehom-closed } \mathbf{B} \ g \ h \ f \ a \ p &= | \ g | ((f \hat{\ } \mathbf{A}) \ a) \equiv \langle | \ g | \ f \ a \rangle \\
&\quad (f \hat{\ } \mathbf{B})(| \ g | \circ a) \equiv \langle \text{ap } (f \hat{\ } \mathbf{B})(gfe \ p) \rangle \\
&\quad (f \hat{\ } \mathbf{B})(| \ h | \circ a) \equiv \langle (| \ h | \ f \ a)^{-1} \rangle \\
&\quad | \ h | ((f \hat{\ } \mathbf{A}) \ a) \blacksquare
\end{aligned}$$

2.1.5 Kernels of Homomorphisms

The *kernel* of a homomorphism is a congruence relation and conversely for every congruence relation θ , there exists a homomorphism with kernel θ (namely, that canonical projection onto the quotient modulo θ).

$$\begin{aligned}
\text{homker-compatible} : \{ \mathbf{A} : \text{Algebra } \mathcal{U} \ S \} \{ \mathbf{B} : \text{Algebra } \mathcal{W} \ S \} (h : \text{hom } \mathbf{A} \ \mathbf{B}) \\
\rightarrow \text{compatible } \mathbf{A} \ (\text{KER-rel } | \ h |)
\end{aligned}$$

$$\text{homker-compatible } \{ \mathbf{A} \} \ \mathbf{B} \ h \ f \ \{ u \} \{ v \} \text{ Kerhab} = \gamma$$

where

$$\begin{aligned}
\gamma : | \ h | ((f \hat{\ } \mathbf{A}) \ u) &\equiv | \ h | ((f \hat{\ } \mathbf{A}) \ v) \\
\gamma &= | \ h | ((f \hat{\ } \mathbf{A}) \ u) \equiv \langle | \ h | \ f \ u \rangle \\
&\quad (f \hat{\ } \mathbf{B})(| \ h | \circ u) \equiv \langle \text{ap } (f \hat{\ } \mathbf{B})(gfe \ \lambda \ x \rightarrow \text{Kerhab } x) \rangle \\
&\quad (f \hat{\ } \mathbf{B})(| \ h | \circ v) \equiv \langle (| \ h | \ f \ v)^{-1} \rangle \\
&\quad | \ h | ((f \hat{\ } \mathbf{A}) \ v) \blacksquare
\end{aligned}$$

$$\begin{aligned}
\text{homker-equivalence} : \{ \mathbf{A} : \text{Algebra } \mathcal{U} \ S \} \{ \mathbf{B} : \text{Algebra } \mathcal{W} \ S \} (h : \text{hom } \mathbf{A} \ \mathbf{B}) \\
\rightarrow \text{IsEquivalence } (\text{KER-rel } | \ h |)
\end{aligned}$$

$$\text{homker-equivalence } \mathbf{A} \ h = \text{map-kernel-IsEquivalence } | \ h |$$

It is convenient to define a function that takes a homomorphism and constructs a congruence from its kernel. We call this function `kercon`.

$$\begin{aligned}
\text{kercon} : \{ \mathbf{A} : \text{Algebra } \mathcal{U} \ S \} \{ \mathbf{B} : \text{Algebra } \mathcal{W} \ S \} (h : \text{hom } \mathbf{A} \ \mathbf{B}) &\rightarrow \text{Congruence } \mathbf{A} \\
\text{kercon } \mathbf{B} \ h &= \text{mkcon } (\text{KER-rel } | \ h |) (\text{homker-compatible } \mathbf{B} \ h) (\text{homker-equivalence } \mathbf{B} \ h)
\end{aligned}$$

With this congruence we define the corresponding quotient, along with some syntactic sugar to denote it.

$$\begin{aligned}
\text{kerquo} : \{ \mathbf{A} : \text{Algebra } \mathcal{U} \ S \} \{ \mathbf{B} : \text{Algebra } \mathcal{W} \ S \} (h : \text{hom } \mathbf{A} \ \mathbf{B}) &\rightarrow \text{Algebra } (\mathcal{U} \sqcup \mathcal{W}^+) \ S \\
\text{kerquo } \{ \mathbf{A} \} \ \mathbf{B} \ h &= \mathbf{A} \ / \ (\text{kercon } \mathbf{B} \ h) \\
\text{[]/ker} : (\mathbf{A} : \text{Algebra } \mathcal{U} \ S) \{ \mathbf{B} : \text{Algebra } \mathcal{W} \ S \} (h : \text{hom } \mathbf{A} \ \mathbf{B}) &\rightarrow \text{Algebra } (\mathcal{U} \sqcup \mathcal{W}^+) \ S \\
\mathbf{A} \ [\ \mathbf{B} \] / \text{ker } h &= \text{kerquo } \{ \mathbf{A} \} \ \mathbf{B} \ h
\end{aligned}$$

Thus, given $h : \text{hom } \mathbf{A} \ \mathbf{B}$, we can construct the quotient of \mathbf{A} modulo the kernel of h , and the UALib syntax for this quotient is $\mathbf{A} \ [\ \mathbf{B} \] / \text{ker } h$.

2.1.6 The natural projection

Given an algebra \mathbf{A} and a congruence θ , the *natural* or *canonical projection* is a map from \mathbf{A} onto $\mathbf{A} \ / \ \theta$ that is constructed, and proved epimorphic, as follows.

$$\begin{aligned}
\pi_{\text{epi}} : \{ \mathbf{A} : \text{Algebra } \mathcal{U} \ S \} \{ \theta : \text{Congruence } \{ \mathcal{U} \} \{ \mathcal{W} \} \ \mathbf{A} \} &\rightarrow \text{epi } \mathbf{A} \ (\mathbf{A} \ / \ \theta) \\
\pi_{\text{epi}} \{ \mathbf{A} \} \ \theta &= c\pi, \text{ c}\pi\text{-is-hom}, \text{ c}\pi\text{-is-epic} \text{ where}
\end{aligned}$$

```

cπ : | A | → | A / θ |
cπ a = [ a ] { ( θ ) }

cπ-is-hom : is-homomorphism A (A / θ) cπ
cπ-is-hom _ _ = refl

cπ-is-epic : Epic cπ
cπ-is-epic (. ( ( θ ) a ) , a , refl) = Image_∃_.im a

```

Sometimes we don't care about the surjectivity of `πepi` and we want to work with its *homomorphic reduct*, which is obtained by applying `epi-to-hom`, like so.

```

πhom : {A : Algebra U S} (θ : Congruence {U} {W} A) → hom A (A / θ)
πhom {A} θ = epi-to-hom (A / θ) (πepi θ)

```

We combine the foregoing to define a function that takes S -algebras \mathbf{A} and \mathbf{B} , and a homomorphism $h : \text{hom } \mathbf{A} \ \mathbf{B}$ and returns the canonical epimorphism from \mathbf{A} onto the quotient algebra $\mathbf{A} [\mathbf{B}] / \ker h$ (which, recall, is \mathbf{A} modulo the kernel of h).

```

πker : {A : Algebra U S} (B : Algebra W S) (h : hom A B) → epi A (A [ B ] / ker h)
πker {A} B h = πepi (kercon B h)

```

The kernel of the canonical projection of \mathbf{A} onto \mathbf{A} / θ is equal to θ , but since equality of inhabitants of certain types (like `Congruence` or `Rel`) can be a tricky business, we settle for proving the containment $\mathbf{A} / \theta \subseteq \theta$. Of the two containments, this is the easier one to prove; luckily it is also the one we need later.

```

ker-in-con : (A : Algebra U S) (θ : Congruence {U} {W} A) (x y : | A |)
→      ( kercon (A / θ) (πhom θ) ) x y → ( θ ) x y

ker-in-con A θ x y hyp = /-refl θ hyp

```

2.1.7 Product homomorphisms

Suppose we have an algebra $\mathbf{A} : \text{Algebra } \mathcal{U} \ S$, a type $I : \mathcal{F} \ \bullet$, and a family $\mathcal{B} : I \rightarrow \text{Algebra } \mathcal{W} \ S$ of algebras. We sometimes refer to the inhabitants of I as *indices*, and call \mathcal{B} an *indexed family of algebras*. If in addition we have a family $\mathcal{h} : (i : I) \rightarrow \text{hom } \mathbf{A} \ (\mathcal{B} \ i)$ of homomorphisms, then we can construct a homomorphism from \mathbf{A} to the product $\prod \mathcal{B}$ in the natural way.

```

∏-hom-co : {A : Algebra U S} {I : F *} (B : I → Algebra W S)
→      ∏ i : I , hom A (B i) → hom A (∏ B)

∏-hom-co {A} B h = φ , φhom
where
φ : | A | → | ∏ B |
φ a = λ i → | h i | a

φhom : is-homomorphism A (∏ B) φ
φhom f a = gfe λ i → || h i || f a

```

The family \mathcal{h} of homomorphisms inhabits the dependent type $\prod i : I , \text{hom } \mathbf{A} \ (\mathcal{B} \ i)$. The syntax we use to represent this type is available to us because of the way `-∏` is defined in the `Type Topology` library. We like this syntax because it is very close to the notation one finds in the standard type theory literature. However, we could equally well have used one of the following

alternatives, which may be closer to “standard Agda” syntax:

$$\prod \lambda i \rightarrow \text{hom } \mathbf{A} (\mathcal{B} i) \quad \text{or} \quad (i : I) \rightarrow \text{hom } \mathbf{A} (\mathcal{B} i) \quad \text{or} \quad \forall i \rightarrow \text{hom } \mathbf{A} (\mathcal{B} i).$$

The foregoing generalizes easily to the case in which the domain is also a product of a family of algebras. That is, if we are given $\mathcal{A} : I \rightarrow \text{Algebra } \mathcal{U} S$ and $\mathcal{B} : I \rightarrow \text{Algebra } \mathcal{W} S$ (two families of S -algebras), and $h : \prod i : I, \text{hom } (\mathcal{A} i) (\mathcal{B} i)$ (a family of homomorphisms), then we can construct a homomorphism from $\prod \mathcal{A}$ to $\prod \mathcal{B}$ in the following natural way.

$$\begin{aligned} \prod\text{-hom} &: \{I : \mathcal{F} \text{ } \cdot\} (\mathcal{A} : I \rightarrow \text{Algebra } \mathcal{U} S) (\mathcal{B} : I \rightarrow \text{Algebra } \mathcal{W} S) \\ &\rightarrow \prod i : I, \text{hom } (\mathcal{A} i) (\mathcal{B} i) \rightarrow \text{hom } (\prod \mathcal{A}) (\prod \mathcal{B}) \end{aligned}$$

$$\prod\text{-hom } \mathcal{A} \mathcal{B} h = \phi, \phi\text{hom}$$

where

$$\phi : | \prod \mathcal{A} | \rightarrow | \prod \mathcal{B} |$$

$$\phi = \lambda x i \rightarrow | h i | (x i)$$

$$\phi\text{hom} : \text{is-homomorphism } (\prod \mathcal{A}) (\prod \mathcal{B}) \phi$$

$$\phi\text{hom } f a = gfe \lambda i \rightarrow \| h i \| f (\lambda x \rightarrow a x i)$$

2.1.8 Projection homomorphisms

Later we will need a proof of the fact that the natural projection out of a product algebra onto one of its factors is a homomorphism.

$$\prod\text{-projection-hom} : \{I : \mathcal{F} \text{ } \cdot\} (\mathcal{B} : I \rightarrow \text{Algebra } \mathcal{W} S) \rightarrow \prod i : I, \text{hom } (\prod \mathcal{B}) (\mathcal{B} i)$$

$$\prod\text{-projection-hom } \mathcal{B} = \lambda i \rightarrow h i, h\text{hom } i$$

where

$$h : \forall i \rightarrow | \prod \mathcal{B} | \rightarrow | \mathcal{B} i |$$

$$h i = \lambda x \rightarrow x i$$

$$h\text{hom} : \forall i \rightarrow \text{is-homomorphism } (\prod \mathcal{B}) (\mathcal{B} i) (h i)$$

$$h\text{hom} _ _ = \text{refl}$$

Of course, we could prove a more general result involving projections onto multiple factors, but so far the single-factor result has sufficed.

2.2 Homomorphism Theorems

This section presents the `Homomorphisms.Noether` module of the `AgdaUALib`, slightly abridged.⁹

2.2.1 The First Homomorphism Theorem

Here is a version of the so-called *First Homomorphism theorem*, sometimes called Noether’s First Homomorphism theorem, after Emmy Noether who was among the first proponents of the abstract approach to algebra that we now refer to as “modern algebra.”

open `Congruence`

module `_` { $\mathcal{U} \mathcal{W} : \text{Universe}$ }

⁹ For unabridged docs and source code see <https://ualib.gitlab.io/Homomorphisms.Noether.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Homomorphisms/Noether.lagda>.

– *extensionality assumptions* –
 $(fe : \text{dfunext } \mathcal{V} \ \mathcal{W}) \ (pe : \text{prop-ext } \mathcal{U} \ \mathcal{W})$

$(\mathbf{A} : \text{Algebra } \mathcal{U} \ S)(\mathbf{B} : \text{Algebra } \mathcal{W} \ S)(h : \text{hom } \mathbf{A} \ \mathbf{B})$

– *truncation assumptions* –
 $(Bset : \text{is-set } |\mathbf{B}|)$
 $(ssR : \forall a \ x \rightarrow \text{is-subsingleton } (\langle \text{kercon } \mathbf{B} \ h \rangle a \ x))$
 $(ssA : \forall C \rightarrow \text{is-subsingleton } (\mathcal{C}\{A = |\mathbf{A}|\} \{ \langle \text{kercon } \mathbf{B} \ h \rangle \} C))$

where

FirstHomomorphismTheorem :

$\Sigma \phi : \text{hom } (\mathbf{A} \ [\mathbf{B}] / \text{ker } h) \ \mathbf{B} ,$
 $(|h| \equiv | \phi \circ | \pi \text{ker } \mathbf{B} \ h |) \times \text{Monic } | \phi | \times \text{is-embedding } | \phi |$

FirstHomomorphismTheorem = $(\phi , \phi \text{hom}) , \phi \text{com} , \phi \text{mon} , \phi \text{emb}$

where

$\theta : \text{Congruence } \mathbf{A}$

$\theta = \text{kercon } \mathbf{B} \ h$

$\phi : |\mathbf{A} \ [\mathbf{B}] / \text{ker } h| \rightarrow |\mathbf{B}|$

$\phi \ a = |h| \ \ulcorner a \urcorner$

$\mathbf{R} : \text{Pred}_2 \ |\mathbf{A}| \ \mathcal{W}$

$\mathbf{R} = \langle \text{kercon } \mathbf{B} \ h \rangle , ssR$

$\phi \text{hom} : \text{is-homomorphism } (\mathbf{A} \ [\mathbf{B}] / \text{ker } h) \ \mathbf{B} \ \phi$

$\phi \text{hom } f \ \mathbf{a} = |h| \ ((f \ \hat{=} \ \mathbf{A}) (\lambda x \rightarrow \ulcorner \mathbf{a} \ x \urcorner)) \equiv \langle ||h|| \ f \ (\lambda x \rightarrow \ulcorner \mathbf{a} \ x \urcorner) \rangle$
 $(f \ \hat{=} \ \mathbf{B}) (|h| \circ (\lambda x \rightarrow \ulcorner \mathbf{a} \ x \urcorner)) \equiv \langle \text{ap } (f \ \hat{=} \ \mathbf{B}) (fe \ \lambda x \rightarrow \text{refl}) \rangle$
 $(f \ \hat{=} \ \mathbf{B}) (\lambda x \rightarrow \phi (\mathbf{a} \ x)) \blacksquare$

$\phi \text{mon} : \text{Monic } \phi$

$\phi \text{mon } (.(\langle \theta \rangle u) , u , \text{refl } _) (.(\langle \theta \rangle v) , v , \text{refl } _) \phi uv =$
 $\text{class-extensionality' } \{ \mathbf{R} = \mathbf{R} \} \ pe \ ssA \ (\text{IsEquiv } \theta) \ \phi uv$

$\phi \text{com} : |h| \equiv \phi \circ | \pi \text{ker } \mathbf{B} \ h |$

$\phi \text{com} = \text{refl}$

$\phi \text{emb} : \text{is-embedding } \phi$

$\phi \text{emb} = \text{monic-is-embedding} | \text{sets } \phi \ Bset \ \phi \text{mon}$

Next we show that the homomorphism ϕ , whose existence we just proved, is unique.

NoetherHomUnique : $(f \ g : \text{hom } (\mathbf{A} \ [\mathbf{B}] / \text{ker } h) \ \mathbf{B})$

$\rightarrow |h| \equiv |f| \circ | \pi \text{ker } \mathbf{B} \ h | \rightarrow |h| \equiv |g| \circ | \pi \text{ker } \mathbf{B} \ h |$

$\rightarrow \forall a \rightarrow |f| \ a \equiv |g| \ a$

NoetherHomUnique $f \ g \ hfk \ h gk \ (.(\langle \text{kercon } \mathbf{B} \ h \rangle a) , a , \text{refl}) =$

$\text{let } \theta = (\langle \text{kercon } \mathbf{B} \ h \rangle a , a , \text{refl}) \text{ in}$

$|f| \ \theta \equiv \langle \text{cong-app } (hfk^{-1}) \ a \rangle \ |h| \ a \equiv \langle \text{cong-app } (h gk) \ a \rangle \ |g| \ \theta \blacksquare$

If we postulate function extensionality, then we have¹⁰

```
fe-NoetherHomUnique : funext (ℳ ⊔ ℳ+) ℳ → (f g : hom (A [ B ]/ker h) B)
→ | h | ≡ | f | ∘ | πker B h | → | h | ≡ | g | ∘ | πker B h |
→ | f | ≡ | g |
```

```
fe-NoetherHomUnique fe f g hfk hgk = fe (NoetherHomUnique f g hfk hgk)
```

Assuming the hypotheses of the First Homomorphism theorem, if we add the assumption that h is epic, then we get the so-called *First Isomorphism theorem*.

FirstIsomorphismTheorem :

```
Epic | h | → Σ f : (epi (A [ B ]/ker h) B) , (| h | ≡ | f | ∘ | πker B h |) × is-embedding | f |
```

```
FirstIsomorphismTheorem hE = (fmap , fhom , fepic) , refl , femb
```

where

```
θ : Congruence A
```

```
θ = kercon B h
```

```
fmap : | A [ B ]/ker h | → | B |
```

```
fmap [a] = | h | ⌈ [a] ⌋
```

```
fhom : is-homomorphism (A [ B ]/ker h) B fmap
```

```
fhom f a = | h | ((f ^ A) λ x → ⌈ a x ⌋) ≡ (| h | f (λ x → ⌈ a x ⌋))
(f ^ B) (| h | ∘ λ x → ⌈ a x ⌋) ≡ (ap (f ^ B) (gfe λ _ → refl))
(f ^ B) (fmap ∘ a) ■
```

```
fepic : Epic fmap
```

```
fepic b = γ where
```

```
a : | A |
```

```
a = EpicInv | h | hE b
```

```
bfa : b ≡ fmap [a]
```

```
bfa = (cong-app (EpicInvsRightInv gfe | h | hE) b)-1
```

```
γ : Image fmap ∋ b
```

```
γ = Image_∋_eq b [a] bfa
```

```
fmon : Monic fmap
```

```
fmon ((⟨ θ ⟩ a) , a , refl) ((⟨ θ ⟩ a') , a' , refl) faa' =
class-extensionality' {R = ⟨ kercon B h ⟩ , ssR} pe ssA (IsEquiv θ) faa'
```

```
femb : is-embedding fmap
```

```
femb = monic-is-embedding|sets fmap Bset fmon
```

The argument used above to prove `NoetherHomUnique` can also be used to prove uniqueness of the epimorphism f found in the isomorphism theorem.

¹⁰We already assumed *global* function extensionality in this module, so we could just appeal to that in this case. However, we make a local function extensionality assumption explicit here merely to highlight where and how the principle is applied.

```

NoetherIsoUnique : (f g : epi (A [ B ]/ker h) B) → | h | ≡ | f | ∘ | πker B h |
→ | h | ≡ | g | ∘ | πker B h | → ∀ a → | f | a ≡ | g | a

NoetherIsoUnique f g hfk hgk (.(< kercon B h > a) , a , refl) =

let θ = (< kercon B h > a , a , refl) in

| f | θ ≡ < cong-app(hfk-1)a > | h | a ≡ < cong-app(hgk)a > | g | θ ■

```

2.2.2 Composition of homomorphisms

The composition of homomorphisms is again a homomorphism. There are a number of alternative ways to formalize this fact in Agda. The two representations included in the [UALib](#) are the following.

```

o-hom : (A : Algebra ℳ S){B : Algebra ℳ S}{C : Algebra ℳ S}
→ hom A B → hom B C → hom A C

o-hom A {B} C (g , ghom) (h , hhom) = h ∘ g , γ where

γ : ∀ f a → (h ∘ g)((f ^ A) a) ≡ (f ^ C)(h ∘ g ∘ a)

γ f a = (h ∘ g)((f ^ A) a) ≡ < ap h ( ghom f a ) >
h ((f ^ B) (g ∘ a)) ≡ < hhom f ( g ∘ a ) >
(f ^ C) (h ∘ g ∘ a) ■

o-is-hom : (A : Algebra ℳ S){B : Algebra ℳ S}{C : Algebra ℳ S}
→ {f : | A | → | B |} {g : | B | → | C |}
→ is-homomorphism A B f → is-homomorphism B C g
→ is-homomorphism A C (g ∘ f)

o-is-hom A C {f} {g} fhom ghom = || o-hom A C (f , fhom) (g , ghom) ||

```

2.2.3 Homomorphism decomposition

If $g : \text{hom } A \ B$, $h : \text{hom } A \ C$, h is surjective, and $\ker h \subseteq \ker g$, then there exists $\phi : \text{hom } C \ B$ such that $g = \phi \circ h$, that is, such that the following diagram commutes.

$$\begin{array}{ccc}
A & \xrightarrow{h} & C \\
& \searrow g & \swarrow \exists \phi \\
& & B
\end{array}$$

This, or some variation of it, is sometimes referred to as the *Second Isomorphism Theorem*. We formalize its statement and proof as follows. (Notice that the proof is constructive.)

```

homFactor : {ℳ : Universe} → funext ℳ ℳ → {A B C : Algebra ℳ S}
→ (g : hom A B) (h : hom A C)
→ ker-pred | h | ⊆ ker-pred | g | → Epic | h |
→ Σ φ : (hom C B) , | g | ≡ | φ | ∘ | h |

```

```

homFactor fe{A}{B}{C}(g , ghom)(h , hhom) Kh⊆Kg hEpi = (ϕ , ϕIsHomCB) , g≡ϕ∘h
where
  hInv : | C | → | A |
  hInv = λ c → (EpicInv h hEpi) c

  ϕ : | C | → | B |
  ϕ = λ c → g ( hInv c )

  ξ : ∀ x → ker-pred h (x , hInv (h x))
  ξ x = (cong-app (EpicInvsRightInv fe h hEpi) (h x))-1

  g≡ϕ∘h : g ≡ ϕ ∘ h
  g≡ϕ∘h = fe λ x → Kh⊆Kg (ξ x)

  ζ : (f : | S |)(c : | S || f → | C |)(x : | S || f) → c x ≡ (h ∘ hInv)(c x)
  ζ f c x = (cong-app (EpicInvsRightInv fe h hEpi) (c x))-1

  υ : (f : | S |)(c : | S || f → | C |) → c ≡ h ∘ (hInv ∘ c)
  υ f c = ap (λ - → - ∘ c)(EpicInvsRightInv fe h hEpi)-1

  useker : ∀ f c → g(hInv (h((f ^ A)(hInv ∘ c)))) ≡ g((f ^ A)(hInv ∘ c))
  useker f c = Kh⊆Kg (cong-app(EpicInvsRightInv fe h hEpi)(h ((f ^ A)(hInv ∘ c))))

  ϕIsHomCB : (f : | S |)(c : | S || f → | C |) → ϕ((f ^ C) c) ≡ (f ^ B)(ϕ ∘ c)
  ϕIsHomCB f c = g (hInv ((f ^ C) c)) ≡⟨ i ⟩
    g (hInv ((f ^ C)(h ∘ (hInv ∘ c)))) ≡⟨ ii ⟩
    g (hInv (h ((f ^ A)(hInv ∘ c)))) ≡⟨ iii ⟩
    g ((f ^ A)(hInv ∘ c)) ≡⟨ iv ⟩
    (f ^ B)(λ x → g (hInv (c x))) ■

  where
    i = ap (g ∘ hInv) (ap (f ^ C) (υ f c))
    ii = ap (g ∘ hInv) (hhom f (hInv ∘ c))-1
    iii = useker f c
    iv = ghom f (hInv ∘ c)

```

Here's a more general version.

```

HomFactor : (A : Algebra ℳ S){B : Algebra ℳ S}{C : Algebra ℳ S}
  (β : hom A B) (γ : hom A C)
→ Epic | γ | → (KER-pred | γ |) ⊆ (KER-pred | β |)
→ Σ ϕ : (hom C B) , | β | ≡ | ϕ ∘ γ |

HomFactor A {B}{C} β γ γE Kγβ = (ϕ , ϕIsHomCB) , βϕγ
where
  γInv : | C | → | A |
  γInv = λ y → (EpicInv | γ | γE) y

  ϕ : | C | → | B |
  ϕ = λ y → | β | ( γInv y )

```

$$\begin{aligned}
& \xi : (x : | \mathbf{A} |) \rightarrow \text{KER-pred } | \gamma | (x, \gamma \text{Inv } (| \gamma | x)) \\
& \xi x = (\text{cong-app } (\text{EpicInvsRightInv } gfe \mid \gamma \mid \gamma E) (| \gamma | x))^{-1} \\
\\
& \beta \phi \gamma : | \beta | \equiv \phi \circ | \gamma | \\
& \beta \phi \gamma = gfe \lambda x \rightarrow K\gamma \beta (\xi x) \\
\\
& \iota : (f : | S |)(c : \| S \| f \rightarrow | \mathbf{C} |) \rightarrow c \equiv | \gamma | \circ (\gamma \text{Inv} \circ c) \\
& \iota f c = \text{ap } (\lambda - \rightarrow - \circ c) (\text{EpicInvsRightInv } gfe \mid \gamma \mid \gamma E)^{-1} \\
\\
& \text{useker} : \forall f c \rightarrow | \beta | (\gamma \text{Inv } (| \gamma | ((f \hat{\ } \mathbf{A}) (\gamma \text{Inv} \circ c)))) \equiv | \beta | ((f \hat{\ } \mathbf{A}) (\gamma \text{Inv} \circ c)) \\
& \text{useker } f c = K\gamma \beta (\text{cong-app } (\text{EpicInvsRightInv } gfe \mid \gamma \mid \gamma E) (| \gamma | ((f \hat{\ } \mathbf{A}) (\gamma \text{Inv} \circ c)))) \\
\\
& \phi \text{IsHomCB} : \forall f c \rightarrow \phi ((f \hat{\ } \mathbf{C}) c) \equiv ((f \hat{\ } \mathbf{B}) (\phi \circ c)) \\
& \phi \text{IsHomCB } f c = | \beta | (\gamma \text{Inv } ((f \hat{\ } \mathbf{C}) c)) \quad \equiv \langle \text{i} \rangle \\
& \quad | \beta | (\gamma \text{Inv } ((f \hat{\ } \mathbf{C}) (| \gamma | \circ (\gamma \text{Inv} \circ c)))) \quad \equiv \langle \text{ii} \rangle \\
& \quad | \beta | (\gamma \text{Inv } (| \gamma | ((f \hat{\ } \mathbf{A}) (\gamma \text{Inv} \circ c)))) \quad \equiv \langle \text{iii} \rangle \\
& \quad | \beta | ((f \hat{\ } \mathbf{A}) (\gamma \text{Inv} \circ c)) \quad \equiv \langle \text{iv} \rangle \\
& \quad ((f \hat{\ } \mathbf{B}) (\lambda x \rightarrow | \beta | (\gamma \text{Inv } (c x)))) \quad \blacksquare \\
\\
& \text{where} \\
& \text{i} = \text{ap } (| \beta | \circ \gamma \text{Inv}) (\text{ap } (f \hat{\ } \mathbf{C}) (\iota f c)) \\
& \text{ii} = \text{ap } (| \beta | \circ \gamma \text{Inv}) (\| \gamma \| f (\gamma \text{Inv} \circ c))^{-1} \\
& \text{iii} = \text{useker } f c \\
& \text{iv} = \| \beta \| f (\gamma \text{Inv} \circ c)
\end{aligned}$$

If, in addition, both β and γ are epic, then so is ϕ .

$$\begin{aligned}
& \text{HomFactorEpi} : (\mathbf{A} : \text{Algebra } \mathcal{X} S) \{ \mathbf{B} : \text{Algebra } \mathcal{Y} S \} \{ \mathbf{C} : \text{Algebra } \mathcal{Z} S \} \\
& \quad (\beta : \text{hom } \mathbf{A} \mathbf{B}) (\beta e : \text{Epic } | \beta |) \\
& \quad (\xi : \text{hom } \mathbf{A} \mathbf{C}) (\xi e : \text{Epic } | \xi |) \\
& \rightarrow (\text{KER-pred } | \xi |) \subseteq (\text{KER-pred } | \beta |) \\
& \rightarrow \Sigma \phi : (\text{epi } \mathbf{C} \mathbf{B}), | \beta | \equiv | \phi | \circ | \xi | \\
\\
& \text{HomFactorEpi } \mathbf{A} \{ \mathbf{B} \} \{ \mathbf{C} \} \beta \beta e \xi \xi e \text{kerincl} = (\text{fst } | \phi F |, (\text{snd } | \phi F |, \phi E)), \| \phi F \| \\
& \text{where} \\
& \phi F : \Sigma \phi : (\text{hom } \mathbf{C} \mathbf{B}), | \beta | \equiv | \phi | \circ | \xi | \\
& \phi F = \text{HomFactor } \mathbf{A} \{ \mathbf{B} \} \{ \mathbf{C} \} \beta \xi \xi e \text{kerincl} \\
\\
& \xi \text{inv} : | \mathbf{C} | \rightarrow | \mathbf{A} | \\
& \xi \text{inv} = \lambda c \rightarrow (\text{EpicInv } | \xi | \xi e) c \\
\\
& \beta \text{inv} : | \mathbf{B} | \rightarrow | \mathbf{A} | \\
& \beta \text{inv} = \lambda b \rightarrow (\text{EpicInv } | \beta | \beta e) b \\
\\
& \phi : | \mathbf{C} | \rightarrow | \mathbf{B} | \\
& \phi = \lambda c \rightarrow | \beta | (\xi \text{inv } c) \\
\\
& \phi E : \text{Epic } \phi \\
& \phi E = \text{epic-factor } gfe \mid \beta \mid | \xi | \phi \| \phi F \| \beta e
\end{aligned}$$

2.3 Isomorphisms

This section presents the `Homomorphisms.Isomorphisms` module of the `AgdaUALib`, slightly abridged.¹¹ Here we formalize the notion of *isomorphism* between algebraic structures.

2.3.1 Definition of isomorphism

Recall, $f \sim g$ means f and g are *extensionally* (or *point-wise*) *equal*; i.e., $\forall x, f x \equiv g x$. We use this notion of equality of functions in the following definition of *isomorphism*.

```
_≅_ : {U W : Universe} {A : Algebra U S} {B : Algebra W S} → O U W U W ·
A ≅ B = Σ f : (hom A B) , Σ g : (hom B A) , (| f | ∘ | g | ∼ | id B |) × (| g | ∘ | f | ∼ | id A |)
```

That is, two structures are *isomorphic* provided there are homomorphisms going back and forth between them which compose to the identity.

2.3.2 Isomorphism is an equivalence relation

```
≅-refl : {U : Universe} {A : Algebra U S} → A ≅ A
≅-refl {U} {A} = id A , id A , (λ a → refl) , (λ a → refl)

≅-sym : {U W : Universe} {A : Algebra U S} {B : Algebra W S} → A ≅ B → B ≅ A
≅-sym h = fst || h || , fst h , || snd || h || , | snd || h || |

≅-trans : {A : Algebra X S} {B : Algebra Y S} {C : Algebra Z S}
→ A ≅ B → B ≅ C → A ≅ C

≅-trans {A} {B} {C} ab bc = f , g , α , β
  where
    f1 : hom A B
    f1 = | ab |
    f2 : hom B C
    f2 = | bc |
    f : hom A C
    f = o-hom A C f1 f2

    g1 : hom C B
    g1 = fst || bc ||
    g2 : hom B A
    g2 = fst || ab ||
    g : hom C A
    g = o-hom C A g1 g2

    α : | f | ∘ | g | ∼ | id C |
    α x = (ap | f2 | (| snd || ab || | (| g1 | x))) · (| snd || bc || |) x

    β : | g | ∘ | f | ∼ | id A |
    β x = (ap | g2 | (| snd || bc || | (| f1 | x))) · (| snd || ab || |) x
```

¹¹For unabridged docs and source code see <https://ualib.gitlab.io/Homomorphisms.Isomorphisms.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Homomorphisms/Isomorphisms.lagda>.

To make `trans-≅` easier to apply in certain situations, we define a couple of alternatives where the only difference is which arguments are implicit.

```

TRANS-≅ : {A : Algebra ℳ S}{B : Algebra ℳ S}{C : Algebra ℳ S}
  →      A ≅ B → B ≅ C → A ≅ C
TRANS-≅ {A}{B}{C} = trans-≅ A B C

Trans-≅ : (A : Algebra ℳ S){B : Algebra ℳ S}{C : Algebra ℳ S}
  →      A ≅ B → B ≅ C → A ≅ C
Trans-≅ A {B} C = trans-≅ A B C

```

2.3.3 Lift is an algebraic invariant

Fortunately, the lift operation preserves isomorphism (i.e., it's an *algebraic invariant*). As algebra is our main focus, this invariance of the lift operation is what makes it a workable solution to the technical problems that arise from the noncumulativity of the universe hierarchy discussed in [Prelude.Lifts](#) [3, §2.5].

```

open Lift

lift-alg-≅ : {A : Algebra ℳ S} → A ≅ (lift-alg A ℳ)
lift-alg-≅ {A} = lift , lower A , extfun lift~lower , extfun (lower~lift{ℳ})

lift-alg-hom : (ℳ : Universe)(ℳ' : Universe){A : Algebra ℳ S}{B : Algebra ℳ' S}
  →      hom A B → hom (lift-alg A ℳ) (lift-alg B ℳ')

lift-alg-hom ℳ ℳ' {A} B (f , fhom) = lift ∘ f ∘ lower , γ
where
  IABh : is-homomorphism (lift-alg A ℳ) B (f ∘ lower)
  IABh = o-is-hom (lift-alg A ℳ) B {lower}{f} (λ _ _ → refl) fhom

  γ : is-homomorphism (lift-alg A ℳ) (lift-alg B ℳ') (lift ∘ (f ∘ lower))
  γ = o-is-hom (lift-alg A ℳ) (lift-alg B ℳ') {f ∘ lower}{lift} IABh λ _ _ → refl

lift-alg-iso : {A : Algebra ℳ S}{ℳ : Universe}{B : Algebra ℳ' S}{ℳ' : Universe}
  →      A ≅ B → (lift-alg A ℳ) ≅ (lift-alg B ℳ')

lift-alg-iso A ≅ B = ≅-trans (≅-trans (≅-sym lift-alg-≅) A ≅ B) lift-alg-≅

```

2.3.4 Lift associativity

The lift is also associative, up to isomorphism at least.

```

lift-alg-assoc : {A : Algebra ℳ S} → lift-alg A (ℳ ⊔ ℳ') ≅ (lift-alg (lift-alg A ℳ) ℳ')
lift-alg-assoc {A} = ≅-trans (≅-trans γ lift-alg-≅) lift-alg-≅
where
  γ : lift-alg A (ℳ ⊔ ℳ') ≅ A
  γ = ≅-sym lift-alg-≅

lift-alg-associative : (A : Algebra ℳ S) → lift-alg A (ℳ ⊔ ℳ') ≅ (lift-alg (lift-alg A ℳ) ℳ')
lift-alg-associative A = lift-alg-assoc {A}

```


2.3.5 Products preserve isomorphisms

Products of isomorphic families of algebras are themselves isomorphic. The proof looks a bit technical, but it is as straightforward as it ought to be.

```

□≅ : {A : I → Algebra U S}{B : I → Algebra W S} → Π i : I , A i ≅ B i → □ A ≅ □ B

□≅ {A}{B} AB = γ
  where
    φ : | □ A | → | □ B |
    φ a i = | fst (AB i) | (a i)

    φhom : is-homomorphism (□ A) (□ B) φ
    φhom f a = gfe (λ i → || fst (AB i) || f (λ x → a x i))

    ψ : | □ B | → | □ A |
    ψ b i = | fst || AB i || | (b i)

    ψhom : is-homomorphism (□ B) (□ A) ψ
    ψhom f b = gfe (λ i → snd | snd (AB i) | f (λ x → b x i))

    φ~ψ : φ ∘ ψ ~ | id (□ B) |
    φ~ψ b = gfe λ i → fst || snd (AB i) || (b i)

    ψ~φ : ψ ∘ φ ~ | id (□ A) |
    ψ~φ a = gfe λ i → snd || snd (AB i) || (a i)

    γ : □ A ≅ □ B
    γ = (φ , φhom) , ((ψ , ψhom) , φ~ψ , ψ~φ)

```

A nearly identical proof goes through for isomorphisms of *lifted* products (though, just for fun, we use the universal quantifier syntax here to express the dependent function type in the statement of the lemma, instead of the Pi notation we used in the statement of the previous lemma; that is, $\forall i \rightarrow A i \cong B (\text{lift } i)$ instead of $\Pi i : I , A i \cong B (\text{lift } i)$).

```

lift-alg-□≅ : {I : F *} {A : I → Algebra U S}{B : (Lift {X} I) → Algebra W S}
  → (∀ i → A i ≅ B (lift i)) → lift-alg (□ A) X ≅ □ B

lift-alg-□≅ {I}{A}{B} AB = γ
  where
    φ : | □ A | → | □ B |
    φ a i = | fst (AB (lower i)) | (a (lower i))

    φhom : is-homomorphism (□ A) (□ B) φ
    φhom f a = gfe (λ i → (|| fst (AB (lower i)) ||) f (λ x → a x (lower i)))

    ψ : | □ B | → | □ A |
    ψ b i = | fst || AB i || | (b (lift i))

    ψhom : is-homomorphism (□ B) (□ A) ψ
    ψhom f b = gfe (λ i → (snd | snd (AB i) |) f (λ x → b x (lift i)))

    φ~ψ : φ ∘ ψ ~ | id (□ B) |
    φ~ψ b = gfe λ i → fst || snd (AB (lower i)) || (b i)

```

$$\begin{aligned} \psi \sim \phi &: \psi \circ \phi \sim | id (\prod \mathcal{A}) | \\ \psi \sim \phi \ a &= gfe \ \lambda \ i \rightarrow \text{snd} \parallel \text{snd} \ (AB \ i) \parallel (a \ i) \end{aligned}$$

$$\begin{aligned} A \cong B &: \prod \mathcal{A} \cong \prod \mathcal{B} \\ A \cong B &= (\phi, \phi \text{hom}), ((\psi, \psi \text{hom}), \phi \sim \psi, \psi \sim \phi) \end{aligned}$$

$$\begin{aligned} \gamma &: \text{lift-alg} (\prod \mathcal{A}) \ \mathcal{I} \cong \prod \mathcal{B} \\ \gamma &= \cong\text{-trans} (\cong\text{-sym lift-alg-}\cong) \ A \cong B \end{aligned}$$

2.3.6 Embedding tools

Finally, we prove some useful facts about embeddings that occasionally come in handy.

$$\begin{aligned} \text{embedding-lift-nat} &: \text{hfunext } \mathcal{I} \ \mathcal{U} \rightarrow \text{hfunext } \mathcal{I} \ \mathcal{W} \\ &\rightarrow \frac{\{I : \mathcal{I} \ \cdot\} \{A : I \rightarrow \mathcal{U} \ \cdot\} \{B : I \rightarrow \mathcal{W} \ \cdot\} \\ &\quad (h : \text{Nat } A \ B) \rightarrow (\forall \ i \rightarrow \text{is-embedding } (h \ i))}{\text{is-embedding}(\text{NatII } h)} \\ &\rightarrow \text{is-embedding}(\text{NatII } h) \end{aligned}$$

$$\text{embedding-lift-nat} \ hfu \ hfw \ h \ hem = \text{NatII-is-embedding} \ hfu \ hfw \ h \ hem$$

$$\begin{aligned} \text{embedding-lift-nat}' &: \text{hfunext } \mathcal{I} \ \mathcal{U} \rightarrow \text{hfunext } \mathcal{I} \ \mathcal{W} \\ &\rightarrow \frac{\{I : \mathcal{I} \ \cdot\} \{\mathcal{A} : I \rightarrow \text{Algebra } \mathcal{U} \ S\} \{\mathcal{B} : I \rightarrow \text{Algebra } \mathcal{W} \ S\} \\ &\quad (h : \text{Nat}(\text{fst} \circ \mathcal{A})(\text{fst} \circ \mathcal{B})) \rightarrow (\forall \ i \rightarrow \text{is-embedding } (h \ i))}{\text{is-embedding}(\text{NatII } h)} \\ &\rightarrow \text{is-embedding}(\text{NatII } h) \end{aligned}$$

$$\text{embedding-lift-nat}' \ hfu \ hfw \ h \ hem = \text{NatII-is-embedding} \ hfu \ hfw \ h \ hem$$

$$\begin{aligned} \text{embedding-lift} &: \text{hfunext } \mathcal{I} \ \mathcal{U} \rightarrow \text{hfunext } \mathcal{I} \ \mathcal{W} \\ &\rightarrow \frac{\{I : \mathcal{I} \ \cdot\} \{\mathcal{A} : I \rightarrow \text{Algebra } \mathcal{U} \ S\} \{\mathcal{B} : I \rightarrow \text{Algebra } \mathcal{W} \ S\} \\ &\quad (h : \forall \ i \rightarrow | \mathcal{A} \ i | \rightarrow | \mathcal{B} \ i |) \rightarrow (\forall \ i \rightarrow \text{is-embedding } (h \ i))}{\text{is-embedding}(\lambda \ (x : | \prod \mathcal{A} |) \ (i : I) \rightarrow (h \ i)(x \ i))} \\ &\rightarrow \text{is-embedding}(\lambda \ (x : | \prod \mathcal{A} |) \ (i : I) \rightarrow (h \ i)(x \ i)) \end{aligned}$$

$$\text{embedding-lift} \ hfu \ hfw \ \{I\} \{\mathcal{A}\} \{\mathcal{B}\} \ h \ hem = \text{embedding-lift-nat}' \ hfu \ hfw \ \{I\} \{\mathcal{A}\} \{\mathcal{B}\} \ h \ hem$$

$$\begin{aligned} \text{iso} \rightarrow \text{embedding} &: \{\mathcal{U} \ \mathcal{W} : \text{Universe}\} \{\mathbf{A} : \text{Algebra } \mathcal{U} \ S\} \{\mathbf{B} : \text{Algebra } \mathcal{W} \ S\} \\ &\rightarrow (\phi : \mathbf{A} \cong \mathbf{B}) \rightarrow \text{is-embedding} (\text{fst} | \phi |) \end{aligned}$$

$$\text{iso} \rightarrow \text{embedding} \ \phi = \text{equivs-are-embeddings} (\text{fst} | \phi |) (\text{invertibles-are-equivs} (\text{fst} | \phi |) \ \text{finv})$$

where

$$\text{finv} : \text{invertible} (\text{fst} | \phi |)$$

$$\text{finv} = | \text{fst} \parallel \phi \parallel |, (\text{snd} \parallel \text{snd } \phi \parallel, \text{fst} \parallel \text{snd } \phi \parallel)$$

2.4 Homomorphic Images

This section presents the `Homomorphisms.HomomorphicImages` module of the `AgdaUALib`, slightly abridged.¹²

We begin with what seems, for our purposes, the most useful way to represent the class of *homomorphic images* of an algebra in dependent type theory.

```
HomImage : {A : Algebra U S} (B : Algebra W S) (ϕ : hom A B) → | B | → U ⊔ W ·
HomImage B ϕ = λ b → Image | ϕ | ⊃ b
```

```
HomImagesOf : Algebra U S → 0 ⊔ V ⊔ U ⊔ W+ ·
HomImagesOf A = Σ B : (Algebra W S) , Σ ϕ : (| A | → | B |) , is-homomorphism A B ϕ × Epic ϕ
```

These types should be self-explanatory, but just to be sure, let's describe the `Sigma` type appearing in the second definition. Given an S -algebra A , the type `HomImagesOf A` denotes the class of algebras $B : \text{Algebra } W \ S$ with a map $\varphi : |A| \rightarrow |B|$ such that φ is an epimorphism.

The standard (informal) notion of the class of homomorphic images of an algebra assumes closure under isomorphism. Thus, we consider B to be a homomorphic image of A if (and only if) there exists an algebra C which is a homomorphic image of A and isomorphic to B . In the `UALib` we express this notion with the following type.

```
_is-hom-image-of_ : (B : Algebra W S) (A : Algebra U S) → ov W ⊔ U ·
B is-hom-image-of A = Σ C ϕ : (HomImagesOf A) , | C ϕ | ≅ B
```

2.4.1 Images of a class of algebras

Given a class \mathcal{K} of S -algebras, we need a type that expresses the assertion that a given algebra is a *homomorphic image* of some algebra in the class, as well as a type that represents all such homomorphic images.

```
_is-hom-image-of-class_ : Algebra U S → Pred (Algebra U S) (U+) → ov U ·
B is-hom-image-of-class K = Σ A : (Algebra U S) , (A ∈ K) × (B is-hom-image-of A)

HomImagesOfClass : Pred (Algebra U S) (U+) → ov U ·
HomImagesOfClass K = Σ B : (Algebra U S) , (B is-hom-image-of-class K)
```

2.4.2 Lifting tools

Here are some tools that have been useful (e.g., in the road to the proof of Birkhoff's HSP theorem). The first states and proves the simple fact that the lift of an epimorphism is an epimorphism.

```
lift-of-alg-epic-is-epic : (X : Universe) {W : Universe}
  {A : Algebra X S} (B : Algebra Y S) (h : hom A B)
```

¹²For unabridged docs and source code see <https://ualib.gitlab.io/Homomorphisms.HomomorphicImages.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Homomorphisms/HomomorphicImages.lagda>.

```

→
Epic | h | → Epic | lift-alg-hom  $\mathcal{X} \mathcal{W} \mathbf{B} h$  |

lift-of-alg-epic-is-epic  $\mathcal{X} \{ \mathcal{W} \} \{ \mathbf{A} \} \mathbf{B} h$  hepi  $y = \text{eq } y (\text{lift } a) \eta$ 
where
lh : hom (lift-alg  $\mathbf{A} \mathcal{X}$ ) (lift-alg  $\mathbf{B} \mathcal{W}$ )
lh = lift-alg-hom  $\mathcal{X} \mathcal{W} \mathbf{B} h$ 

 $\zeta$  : Image | h |  $\ni$  (lower  $y$ )
 $\zeta = \text{hepi } (\text{lower } y)$ 

a : |  $\mathbf{A}$  |
a = Inv | h |  $\zeta$ 

 $\beta$  : lift (| h | a)  $\equiv$  (lift  $\circ$  | h |  $\circ$  lower  $\{ \mathcal{W} \}$ ) (lift a)
 $\beta = \text{ap } (\lambda - \rightarrow \text{lift } (| h | (- a))) (\text{lower} \sim \text{lift } \{ \mathcal{W} \})$ 

 $\eta$  :  $y \equiv$  | lh | (lift a)
 $\eta = y \equiv$  (extfun lift  $\sim$  lower)  $y$ 
lift (lower  $y$ )  $\equiv$  ap lift (InvInv | h |  $\zeta$ )-1
lift (| h | a)  $\equiv$   $\beta$ 
| lh | (lift a) ■

lift-alg-hom-image : {  $\mathcal{X} \mathcal{W}$  : Universe }
{  $\mathbf{A}$  : Algebra  $\mathcal{X} S$  } {  $\mathbf{B}$  : Algebra  $\mathcal{Y} S$  }
→
B is-hom-image-of A

→
(lift-alg  $\mathbf{B} \mathcal{W}$ ) is-hom-image-of (lift-alg  $\mathbf{A} \mathcal{X}$ )

lift-alg-hom-image {  $\mathcal{X}$  } {  $\mathcal{W}$  } {  $\mathbf{A}$  } {  $\mathbf{B}$  } (( $\mathbf{C}$  ,  $\phi$  ,  $\phi \text{hom}$  ,  $\phi \text{epic}$ ) ,  $C \cong B$ ) =
(lift-alg  $\mathbf{C} \mathcal{W}$  , |  $\phi$  | , ||  $\phi$  || ,  $\phi \text{epic}$ ) , lift-alg-iso  $C \cong B$ 
where
 $\phi$  : hom (lift-alg  $\mathbf{A} \mathcal{X}$ ) (lift-alg  $\mathbf{C} \mathcal{W}$ )
 $\phi = (\text{lift-alg-hom } \mathcal{X} \mathcal{W} \mathbf{C}) (\phi , \phi \text{hom})$ 

 $\phi \text{epic}$  : Epic |  $\phi$  |
 $\phi \text{epic} = \text{lift-of-alg-epic-is-epic } \mathcal{X} \mathbf{C} (\phi , \phi \text{hom}) \phi \text{epic}$ 

```

3 Types for Terms

3.1 Basic Definitions

This section presents the `Terms.Basic` module of the `AgdaUALib`, slightly abridged.¹³ The theoretical background that begins each subsection below is based on Section 4.3 of Cliff Bergman's excellent textbook on universal algebra, [1, §4.3]. Apart from notation, our presentation is similar to Bergman's, but we will try to be concise, omitting some details and examples, in order to more quickly arrive at our objective, which is to use type theory to express the concepts and formalize them in the `Agda` language. We refer the reader to [1] for a more complete exposition of classical (informal) universal algebra.

¹³For unabridged docs and source code see <https://ualib.gitlab.io/Terms.Basic.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Terms/Basic.lagda>.

3.1.1 The type of terms

Fix a signature S and let X denote an arbitrary nonempty collection of variable symbols. Assume the symbols in X are distinct from the operation symbols of S , that is $X \cap |S| = \emptyset$. By a *word* in the language of S , we mean a nonempty, finite sequence of members of $X \cup |S|$. We denote the concatenation of such sequences by simple juxtaposition. Let S_0 denote the set of nullary operation symbols of S . We define by induction on n the sets T_n of *words* over $X \cup |S|$ as follows (cf. [1, Def. 4.19]):

$$T_0 := X \cup S_0 \text{ and } T_{n+1} := T_n \cup \mathcal{T}_n,$$

where \mathcal{T}_n is the collection of all $f \ t$ such that $f : |S|$ and $t : \|S\| f \rightarrow T_n$. (Recall, $\|S\| f$ denotes the arity of the operation symbol f .)

We define the collection of *terms* in the signature S over X by $\text{Term } X := \bigcup_n T_n$. By an *S-term* we mean a term in the language of S . Since the definition of $\text{Term } X$ is recursive, it would seem that an inductive type could be used to represent the semantic notion of terms in type theory. Indeed, the *inductive type of terms* in the UALib is one such representation; it is defined as follows.

```
data Term {X : Universe} (X : X → •) : ov X → • where
  generator : X → Term X
  node      : (f : |S|) (t : \|S\| f → Term X) → Term X
```

Notation. As usual, the type X represents an arbitrary collection of variable symbols. Recall, $\text{ov } X$ is our shorthand notation for the universe $\mathbb{O} \sqcup \mathcal{V} \sqcup X^+$. Throughout this module the name of the first constructor of the Term type will remain *generator*. However, in all of the modules that follow this one, we will use the shorthand *g* to denote the *generator* constructor.

3.1.2 The term algebra

For a given signature S , if the type $\text{Term } X$ is nonempty (equivalently, if X or $|S|$ is nonempty), then we can define an algebraic structure, denoted by $\mathbf{T } X$, called the *term algebra in the signature S over X* . Since terms take other terms as arguments they do double-duty, serving as both the elements of the domain and the basic operations of the algebra.

- For each operation symbol $f : |S|$, denote by $f^\wedge(\mathbf{T } X)$ the operation on $\text{Term } X$ which maps each tuple $t : \|S\| f \rightarrow \mathbf{T } X$ of terms to the formal term $f \ t$.
- Define $\mathbf{T } X$ to be the algebra with universe $|\mathbf{T } X| := \text{Term } X$ and operations $f^\wedge(\mathbf{T } X)$, one for each symbol f in $|S|$.

In Agda the term algebra can be defined as simply as one could hope.

```
T : {X : Universe} (X : X → •) → Algebra (ov X) S
T X = Term X , node
```

3.1.3 The universal property

The term algebra $\mathbf{T } X$ is *absolutely free* (or *universal* or *initial*) for algebras in the signature S . That is, for every S -algebra \mathbf{A} , the following hold.

1. Every function from X to $|\mathbf{A}|$ lifts to a homomorphism from $\mathbf{T } X$ to \mathbf{A} .
2. The homomorphism that exists by item 1 is unique.

We now prove this in [Agda](#), starting with the fact that every map from X to $| \mathbf{A} |$ lifts to a map from $| \mathbf{T} X |$ to $| \mathbf{A} |$ in a natural way, by induction on the structure of a given term.

```

free-lift : (A : Algebra U S) (h : X → | A |) → | T X | → | A |
free-lift _ h (generator x) = h x
free-lift A h (node f t) = (f ^ A) (λ i → free-lift A h (t i))

```

Naturally, at the base step of the induction, when the term has the form `generator x`, the free lift of h agrees with h . For the inductive step, when the given term has the form `node f t`, the free lift is defined as follows: Assuming (the induction hypothesis) that we know the image of each subterm $t i$ under the free lift of h , define the free lift at the full term by applying $f ^ A$ to the images of the subterms.

The free lift so defined is a homomorphism by construction. Indeed, here is the formal statement and (trivial) proof of this fact.

```

lift-hom : (A : Algebra U S) → (X → | A |) → hom (T X) A
lift-hom A h = free-lift A h , λ f a → ap (f ^ A) refl

```

Finally, we prove that the homomorphism is unique. This requires `funext` $\mathcal{V} \mathcal{U}$ (i.e., *function extensionality* at universe levels \mathcal{V} and \mathcal{U}) which we postulate by making it part of the premise in the following function type definition.

```

free-unique : funext V U → (A : Algebra U S) (g h : hom (T X) A)
→      (∀ x → | g | (generator x) ≡ | h | (generator x))
-----
→      ∀ (t : Term X) → | g | t ≡ | h | t

free-unique _ _ _ p (generator x) = p x
free-unique fe A g h p (node f t) = | g | (node f t) ≡ ⟨ || g || f t ⟩
                                   (f ^ A) (| g | ∘ t) ≡ ⟨ α ⟩
                                   (f ^ A) (| h | ∘ t) ≡ ⟨ (|| h || f t)-1 ⟩
                                   | h | (node f t) ■

where
α : (f ^ A) (| g | ∘ t) ≡ (f ^ A) (| h | ∘ t)
α = ap (f ^ A) (fe λ i → free-unique fe A g h p (t i))

```

Let's account for what we have proved thus far about the term algebra. If we postulate a type $X : \mathcal{X}$ (representing an arbitrary collection of variable symbols) such that for each S -algebra \mathbf{A} there is a map from X to the domain of \mathbf{A} , then it follows that for every S -algebra \mathbf{A} there is a homomorphism from $\mathbf{T} X$ to $| \mathbf{A} |$ that “agrees with the original map on X ,” by which we mean that for all $x : X$ the lift evaluated at `generator x` is equal to the original function evaluated at x .

If we further assume that each of the mappings from X to $| \mathbf{A} |$ is *surjective*, then the homomorphisms constructed with `free-lift` and `lift-hom` are *epimorphisms*, as we now prove.

```

lift-of-epi-is-epi : {A : Algebra U S} {h₀ : X → | A |}
-----
→      Epic h₀ → Epic | lift-hom A h₀ |

lift-of-epi-is-epi {A} {h₀} hE y = γ
where
h₀-1 y = Inv h₀ (hE y)

```

$$\begin{aligned}\eta : y &\equiv \mid \text{lift-hom } \mathbf{A} \ h_0 \mid (\text{generator } h_0^{-1} y) \\ \eta &= (\text{InvIsInv } h_0 \ (hE \ y))^{-1} \\ \gamma : \text{Image} \mid \text{lift-hom } \mathbf{A} \ h_0 \mid \ni y \\ \gamma &= \text{eq } y \ (\text{generator } h_0^{-1} y) \ \eta\end{aligned}$$

The `lift-hom` and `lift-of-epi-is-epi` types will be called to action when such epimorphisms are needed later (e.g., in the `Varieties` module).

3.2 Term Operations

This section presents the `Terms.Operations` module of the `AgdaUALib`, slightly abridged.¹⁴ Here we define *term operations* which are simply terms interpreted in a particular algebra, and we prove some compatibility properties of term operations.¹⁵

When we interpret a term in an algebra we call the resulting function a *term operation*. Given a term p and an algebra \mathbf{A} , we denote by $p \cdot \mathbf{A}$ the *interpretation* of p in \mathbf{A} . This is defined inductively as follows.

1. If $p = x$ (a variable symbol) and $a : X \rightarrow \mid \mathbf{A} \mid$ a tuple of elements from the domain of \mathbf{A} , then $(p \cdot \mathbf{A}) \ a := a \ x$.
2. If $p = f \ t$ (where f is an operation symbol and t is a tuple of terms) and if $a : X \rightarrow \mid \mathbf{A} \mid$ is a tuple from \mathbf{A} , then we define $(p \cdot \mathbf{A}) \ a = (f \ ^\wedge \ \mathbf{A}) \ (\lambda \ i \rightarrow (t \ i \cdot \mathbf{A}) \ a)$.

Thus the interpretation of a term is defined by induction on the structure of the term, and the definition is formally implemented in the `UALib` as follows.

$$\begin{aligned}_ \cdot _ : \text{Term } X \rightarrow (\mathbf{A} : \text{Algebra } \mathcal{U} \ S) \rightarrow (X \rightarrow \mid \mathbf{A} \mid) \rightarrow \mid \mathbf{A} \mid \\ (\text{g } x \cdot \mathbf{A}) \ a = a \ x \\ (\text{node } f \ t \cdot \mathbf{A}) \ a = (f \ ^\wedge \ \mathbf{A}) \ \lambda \ i \rightarrow (t \ i \cdot \mathbf{A}) \ a\end{aligned}$$

It turns out that the interpretation of a term is the same as the `free-lift` (modulo argument order and assuming function extensionality).

$$\begin{aligned}\text{free-lift-interp} : \text{dfunext } \mathcal{V} \ \mathcal{U} \rightarrow (\mathbf{A} : \text{Algebra } \mathcal{U} \ S)(h : X \rightarrow \mid \mathbf{A} \mid)(p : \text{Term } X) \\ \rightarrow (p \cdot \mathbf{A}) \ h \equiv (\text{free-lift } \mathbf{A} \ h) \ p \\ \text{free-lift-interp } _ \ \mathbf{A} \ h \ (\text{g } x) = \text{refl} \\ \text{free-lift-interp } fe \ \mathbf{A} \ h \ (\text{node } f \ t) = \text{ap } (f \ ^\wedge \ \mathbf{A}) \ (fe \ \lambda \ i \rightarrow \text{free-lift-interp } fe \ \mathbf{A} \ h \ (t \ i))\end{aligned}$$

If the algebra \mathbf{A} happens to be $\mathbf{T} \ X$, then we expect that $\forall s$ we have $(p \cdot \mathbf{T} \ X) \ s \equiv p \ s$. But what is $(p \cdot \mathbf{T} \ X) \ s$ exactly? By definition, it depends on the form of p as follows:

- if $p \equiv \text{g } x$, then $(p \cdot \mathbf{T} \ X) \ s := (\text{g } x \cdot \mathbf{T} \ X) \ s \equiv s \ x$;
- if $p = \text{node } f \ t$, then $(p \cdot \mathbf{T} \ X) \ s := (\text{node } f \ t \cdot \mathbf{T} \ X) \ s \equiv (f \ ^\wedge \ \mathbf{T} \ X) \ \lambda \ i \rightarrow (t \ i \cdot \mathbf{T} \ X) \ s$.

Now, assume $\phi : \text{hom } \mathbf{T} \ \mathbf{A}$. Then by `comm-hom-term`, we have $\mid \phi \mid (p \cdot \mathbf{T} \ X) \ s \equiv (p \cdot \mathbf{A}) \mid \phi \mid \circ s$.

¹⁴For unabridged docs and source code see <https://ualib.gitlab.io/Terms.Operations.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Terms/Operations.lagda>.

¹⁵**Notation.** At the start of the `TermsOperations` module, for notational convenience we rename the `generator` constructor of the `Term` type, so that from now on we can use `g` in place of `generator`.

■ if $p \equiv \mathbf{g} \ x$ (and $t : X \rightarrow | \mathbf{T} \ X |$), then

$$| \phi | \ p := | \phi | \ (\mathbf{g} \ x) \equiv | \phi | \ (\lambda t \rightarrow t \ x) \equiv \lambda t \rightarrow (| \phi | \circ t) \ x;$$

■ if $p \equiv \mathbf{node} \ f \ t$, then

$$| \phi | \ p := | \phi | \ (p \cdot \mathbf{T} \ X) \ s = (\mathbf{node} \ f \ t \cdot \mathbf{T} \ X) \ s \ (f \hat{\ } \mathbf{T} \ X) \ \lambda i \rightarrow (t \ i \cdot \mathbf{T} \ X) \ s.$$

We claim that for all $p : \mathbf{Term} \ X$ there exist $q : \mathbf{Term} \ X$ and $t : X \rightarrow | \mathbf{T} \ X |$ such that $p \equiv (q \cdot \mathbf{T} \ X) \ t$. We prove this fact as follows.

$$\begin{aligned} \text{term-interp} &: \{ \mathcal{X} : \mathbf{Universe} \} \{ X : \mathcal{X} \cdot \} \{ f : | S | \} \{ s t : || S || \ f \rightarrow \mathbf{Term} \ X \} \\ &\rightarrow s \equiv t \rightarrow \mathbf{node} \ f \ s \equiv (f \hat{\ } \mathbf{T} \ X) \ t \end{aligned}$$

$$\text{term-interp} \ f \ \{ s \} \{ t \} \ st = \mathbf{ap} \ (\mathbf{node} \ f) \ st$$

module $_$ $\{ \mathcal{X} : \mathbf{Universe} \} \{ X : \mathcal{X} \cdot \} \{ fe : \mathbf{dfunext} \ \mathcal{V} \ (\mathbf{ov} \ \mathcal{X}) \}$ **where**

$$\text{term-gen} : (p : | \mathbf{T} \ X |) \rightarrow \Sigma \ q : | \mathbf{T} \ X |, \ p \equiv (q \cdot \mathbf{T} \ X) \ \mathbf{g}$$

$$\text{term-gen} \ (\mathbf{g} \ x) = (\mathbf{g} \ x), \ \mathbf{refl}$$

$$\text{term-gen} \ (\mathbf{node} \ f \ t) = \mathbf{node} \ f \ (\lambda i \rightarrow | \text{term-gen} \ (t \ i) |), \ \text{term-interp} \ f \ (fe \ \lambda i \rightarrow || \text{term-gen} \ (t \ i) ||)$$

$$\text{term-gen-agreement} : (p : | \mathbf{T} \ X |) \rightarrow (p \cdot \mathbf{T} \ X) \ \mathbf{g} \equiv (| \text{term-gen} \ p | \cdot \mathbf{T} \ X) \ \mathbf{g}$$

$$\text{term-gen-agreement} \ (\mathbf{g} \ x) = \mathbf{refl}$$

$$\text{term-gen-agreement} \ (\mathbf{node} \ f \ t) = \mathbf{ap} \ (f \hat{\ } \mathbf{T} \ X) \ (fe \ \lambda x \rightarrow \text{term-gen-agreement} \ (t \ x))$$

$$\text{term-agreement} : (p : | \mathbf{T} \ X |) \rightarrow p \equiv (p \cdot \mathbf{T} \ X) \ \mathbf{g}$$

$$\text{term-agreement} \ p = \mathbf{snd} \ (\text{term-gen} \ p) \cdot (\text{term-gen-agreement} \ p)^{-1}$$

3.2.1 Interpretation of terms in product algebras

Note that while in the previous section it sufficed to postulate a local version of function extensionality, in the present section we will assume the full global version ([global-dfunext](#)) is in force. (We are not sure whether this is necessary or if, with some effort, we could get a more moderate invocation of function extensionality to work here.)¹⁶

$$\begin{aligned} \text{interp-prod} &: \{ \mathcal{W} : \mathbf{Universe} \} \{ p : \mathbf{Term} \ X \} \{ I : \mathcal{W} \cdot \} \\ &\quad (\mathcal{A} : I \rightarrow \mathbf{Algebra} \ \mathcal{U} \ S) (a : X \rightarrow \forall i \rightarrow | \mathcal{A} \ i |) \end{aligned}$$

$$\rightarrow (p \cdot (\prod \mathcal{A})) \ a \equiv (\lambda i \rightarrow (p \cdot \mathcal{A} \ i) \ (\lambda j \rightarrow a \ j \ i))$$

$$\text{interp-prod} \ (\mathbf{g} \ x_1) \ \mathcal{A} \ a = \mathbf{refl}$$

$$\text{interp-prod} \ (\mathbf{node} \ f \ t) \ \mathcal{A} \ a = \mathbf{let} \ IH = \lambda x \rightarrow \text{interp-prod} \ (t \ x) \ \mathcal{A} \ a$$

in

$$(f \hat{\ } \prod \mathcal{A}) (\lambda x \rightarrow (t \ x \cdot \prod \mathcal{A}) \ a) \equiv \langle \mathbf{ap} \ (f \hat{\ } \prod \mathcal{A}) (gfe \ IH) \rangle$$

$$(f \hat{\ } \prod \mathcal{A}) (\lambda x \rightarrow (\lambda i \rightarrow (t \ x \cdot \mathcal{A} \ i) (\lambda j \rightarrow a \ j \ i))) \equiv \langle \mathbf{refl} \rangle$$

$$(\lambda i \rightarrow (f \hat{\ } \mathcal{A} \ i) (\lambda x \rightarrow (t \ x \cdot \mathcal{A} \ i) (\lambda j \rightarrow a \ j \ i))) \ \blacksquare$$

$$\text{interp-prod2} : (p : \mathbf{Term} \ X) \{ I : \mathcal{U} \cdot \} \{ \mathcal{A} : I \rightarrow \mathbf{Algebra} \ \mathcal{U} \ S \}$$

$$\rightarrow (p \cdot \prod \mathcal{A}) \equiv \lambda (t : X \rightarrow | \prod \mathcal{A} |) \rightarrow (\lambda i \rightarrow (p \cdot \mathcal{A} \ i) (\lambda x \rightarrow t \ x \ i))$$

¹⁶We plan to resolve this, and if possible improve upon our treatment of function extensionality, before the next major release of the [AgdaUALib](#).


```

interp-prod2 (g x1) A = refl
interp-prod2 (node f t) A = gfe λ (tup : X → | ∏ A |) →
  let IH = λ x → interp-prod (t x) A in
  let tA = λ z → t z · ∏ A in
  (f ^ ∏ A)(λ s → tA s tup) ≡⟨ ap(f ^ ∏ A)(gfe λ x → IH x tup) ⟩
  (f ^ ∏ A)(λ s → λ j → (t s · A j)(λ ℓ → tup ℓ j)) ≡⟨ refl ⟩
  (λ i → (f ^ A i)(λ s → (t s · A i)(λ ℓ → tup ℓ i))) ■

```

3.2.2 Compatibility of terms

We now prove two important facts about term operations. The first of these, which is used very often in the sequel, asserts that every term commutes with every homomorphism.

```

comm-hom-term : {A : Algebra U S} (B : Algebra W S)
  (h : hom A B) (t : Term X) (a : X → | A |)
  → | h | ((t · A) a) ≡ (t · B) (| h | ∘ a)

comm-hom-term B h (g x) a = refl
comm-hom-term {A} B h (node f t) a = | h | ((f ^ A) λ i → (t i · A) a) ≡⟨ i ⟩
  (f ^ B)(λ i → | h | ((t i · A) a)) ≡⟨ ii ⟩
  (f ^ B)(λ r → (t r · B)(| h | ∘ a)) ■

where
i = | h | f(λ r → (t r · A) a)
ii = ap (f ^ B)(gfe (λ i → comm-hom-term B h (t i) a))

```

To conclude the `Terms` module, we prove that every term is compatible with every congruence relation. That is, if $t : \text{Term } X$ and $\theta : \text{Con } A$, then $a \theta b \rightarrow t(a) \theta t(b)$.

```

open Congruence
compatible-term : {A : Algebra U S} (t : Term X) (θ : Con A)
  → compatible-fun (t · A) | θ |

compatible-term (g x) θ p = p x
compatible-term (node f t) θ p = snd || θ || f λ x → (compatible-term (t x) θ) p

```

4 Subalgebra Types

4.1 Subuniverses

This section presents the `Subalgebras.Subuniverses` module of the `AgdaUALib`, slightly abridged.¹⁷ We start by defining a type that represents the important concept of *subuniverse*. Suppose A is an algebra. A subset $B \subseteq | A |$ is said to be *closed under the operations of A* if for each $f \in | S |$ and all tuples $b : || S || f \rightarrow B$ the element $(f ^ A) b$ belongs to B . If a subset $B \subseteq A$ is closed under the operations of A , then we call B a *subuniverse* of A .

¹⁷For unabridged docs and source code see <https://ualib.gitlab.io/Subalgebras.Subuniverses.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Subalgebras/Subuniverses.lagda>.

We first show how to represent in [Agda](#) the collection of subuniverses of an algebra \mathbf{A} . Since a subuniverse is viewed as a subset of the domain of \mathbf{A} , we define it as a predicate on $|\mathbf{A}|$. Thus, the collection of subuniverses is a predicate on predicates on $|\mathbf{A}|$.

```
Subuniverses : (A : Algebra U S) → Pred (Pred | A | W) (⊆ ⊔ ∩ ⊔ U ⊔ W)
Subuniverses A B = (f : | S |)(a : || S || f → | A |) → Im a ⊆ B → (f ^ A) a ∈ B
```

An algebra can be constructed out of a subuniverse in the following natural way.

```
SubunivAlg : (A : Algebra U S)(B : Pred | A | W) → B ∈ Subuniverses A → Algebra (U ⊔ W) S
SubunivAlg A B B ∈ SubA = Σ B , λ f b → (f ^ A)(fst o b) , B ∈ SubA f (fst o b)(snd o b)
```

4.1.1 Subuniverses as records

Next we define a type to represent a single subuniverse of an algebra. If \mathbf{A} is the algebra in question, then the subuniverse will be a subset of (i.e., predicate over) the domain $|\mathbf{A}|$ that belongs to `Subuniverses A`.

```
record Subuniverse {A : Algebra U S} : ov (U ⊔ W) · where
  constructor mksub
  field
    sset : Pred | A | W
    isSub : sset ∈ Subuniverses A
```

As an example application, here is a formal proof that the equalizer of two homomorphisms with domain \mathbf{A} is a subuniverse of \mathbf{A} .

```
Ehom-is-subuniverse : dfunext V W → {A : Algebra U S}(B : Algebra W S)(g h : hom A B)
→ Subuniverse {A = A}

Ehom-is-subuniverse fe B g h =
  mksub (Ehom{fe = fe} B g h) λ f a x → Ehom-closed{fe = fe} B g h f a x
```

4.1.2 Subuniverse Generation

If \mathbf{A} is an algebra and $X ⊆ |\mathbf{A}|$ a subset of the domain of \mathbf{A} , then the *subuniverse of \mathbf{A} generated by X* is typically denoted by $\mathbf{Sg}^{\mathbf{A}}(X)$ and defined to be the smallest subuniverse of \mathbf{A} containing X . Equivalently,

$$\mathbf{Sg}^{\mathbf{A}}(X) = \bigcap \{U : U \text{ is a subuniverse of } \mathbf{A} \text{ and } X ⊆ U\}.$$

We define an inductive type, denoted by `Sg`, that represents the subuniverse generated by a given subset of the domain of a given algebra, as follows.

```
data Sg (A : Algebra U S)(X : Pred | A | W) : Pred | A | (⊆ ⊔ ∩ ⊔ W ⊔ U) where
  var : ∀ {v} → v ∈ X → v ∈ Sg A X
  app : (f : | S |)(a : || S || f → | A |) → Im a ⊆ Sg A X → (f ^ A) a ∈ Sg A X
```

Given an arbitrary S -algebra \mathbf{A} and subset X of $|\mathbf{A}|$, the type `Sg X` does indeed represent a subuniverse of \mathbf{A} ; proving this with the inductive type `Sg` is trivial, as we see here.

```
sgIsSub : {A : Algebra U S}{X : Pred | A | W} → Sg A X ∈ Subuniverses A
sgIsSub = app
```

Next we prove by structural induction that `Sg X` is the smallest subuniverse of \mathbf{A} containing X .

```

sglsSmallest : {ℛ : Universe}{A : Algebra ℳ S}{X : Pred | A | ℳ}{Y : Pred | A | ℛ}
→ Y ∈ Subuniverses A → X ⊆ Y → Sg A X ⊆ Y

sglsSmallest _ _ _ XinY (var Xv) = XinY Xv

sglsSmallest A Y YsubA XinY (app f a SgXa) = fa∈Y
  where
    IH : Im a ⊆ Y
    IH i = sglsSmallest A Y YsubA XinY (SgXa i)

    fa∈Y : (f ^ A) a ∈ Y
    fa∈Y = YsubA f a IH

```

When the inhabitant of `Sg X` is constructed as `app f a SgXa`, we may assume (the induction hypothesis) that the arguments in the tuple `a` belong to `Y`. Then the result of applying `f` to `a` also belongs to `Y` since `Y` is a subuniverse.

4.1.3 Subuniverse Lemmas

Here we formalize a few basic properties of subuniverses. First, the intersection of subuniverses is again a subuniverse.

```

sub-intersection : {A : Algebra ℳ S}{I : ℱ ·}{ℳ : I → Pred | A | ℳ}
→ (Π i : I, ℳ i ∈ Subuniverses A) → ⋂ I ℳ ∈ Subuniverses A

sub-intersection α f a β = λ i → α i f a λ j → β j i

```

In the proof above, we assume the following typing judgments:

```

α : ∀ i → ℳ i ∈ Subuniverses A
f : | S |
a : || S || f → | A |
β : Im a ⊆ ⋂ I ℳ

```

and we must prove $(f \text{ } ^{\text{A}} \text{ } a) \in \bigcap I \mathcal{M}$. In this case, Agda will fill in the proof term $\lambda i \rightarrow \alpha i f a (\lambda x \rightarrow \beta x i)$ automatically with the command `C-c C-a` in `agda2-mode`.

Next we formalize the proof that subuniverses are closed under the action of term operations.

```

sub-term-closed : {ℳ : Universe}{X : ℳ ·}{A : Algebra ℳ S}{B : Pred | A | ℳ}
→ (B ∈ Subuniverses A) → (t : Term X)(b : X → | A |)
→ (∀ x → b x ∈ B) → ((t · A) b) ∈ B

sub-term-closed A α (g x) b Bb = Bb x
sub-term-closed A {B} α (node f t) b β =
  α f (λ z → (t z · A) b) λ x → sub-term-closed A {B} α (t x) b β

```

In the induction step of the foregoing proof, the typing judgments of the premise are these:

```

A : Algebra ℳ S    B : Pred | A | ℳ    α : B ∈ Subuniverses A
f : | S |          t : || S || f → Term X
b : X → | A |     β : ∀ x → b x ∈ B

```

This is another instance in which Agda will fill in the correct proof term if we invoke the command `C-c C-a` in `agda2-mode`.

Alternatively, we could express the preceding fact using an inductive type representing images of terms.

```

data TermImage (A : Algebra U S)(Y : Pred | A | W) : Pred | A | (O ⊔ V' ⊔ U ⊔ W)
where
var : ∀ {y : | A |} → y ∈ Y → y ∈ TermImage A Y
app : ∀ f t → Π x : || S || f , t x ∈ TermImage A Y → (f ^ A) t ∈ TermImage A Y

```

By what we proved above, it should come as no surprise that $\text{TermImage } A \ Y$ is a subuniverse of A that contains Y . Indeed, the proof is trivial.

```

TermImagesSub : {A : Algebra U S}{Y : Pred | A | W} → TermImage A Y ∈ Subuniverses A
TermImagesSub = app

```

```

Y-onlyif-TermImageY : {A : Algebra U S}{Y : Pred | A | W} → Y ⊆ TermImage A Y
Y-onlyif-TermImageY {a} Ya = var Ya

```

Since $\text{Sg } A \ Y$ is the smallest subuniverse containing Y , we obtain the following inclusion.

```

SgY-onlyif-TermImageY : (A : Algebra U S)(Y : Pred | A | W) → Sg A Y ⊆ TermImage A Y
SgY-onlyif-TermImageY A Y = sglsSmallest A (TermImage A Y) TermImagesSub Y-onlyif-TermImageY

```

4.1.4 Homomorphic images are subuniverses

Now that we have developed the machinery of subuniverse generation, we can prove two basic facts that play an important role in many theorems about algebraic structures. First, the image of a homomorphism is a subuniverse of its codomain.

```

hom-image-is-sub : {A : Algebra U S}{B : Algebra W S}
  (φ : hom A B) → (HomImage B φ) ∈ Subuniverses B

```

```

hom-image-is-sub {A}{B} φ f b Imfb = eq ((f ^ B) b) ((f ^ A) ar) γ

```

where

```

ar : || S || f → | A |
ar = λ x → Inv | φ | (Imfb x)

```

```

ζ : | φ | ∘ ar ≡ b
ζ = gfe (λ x → InvlInv | φ | (Imfb x))

```

```

γ : (f ^ B) b ≡ | φ | ((f ^ A) ar)
γ = (f ^ B) b ≡ ⟨ ap (f ^ B)(ζ-1) ⟩
  (f ^ B) (| φ | ∘ ar) ≡ ⟨ (|| φ || f ar)-1 ⟩
  | φ | ((f ^ A) ar) ■

```

Next we prove the important fact that homomorphisms are uniquely determined by their values on a generating set.

```

hom-unique : funext W W → {A : Algebra U S}{B : Algebra W S}
  (X : Pred | A | U) (g h : hom A B)
→
  Π x : | A | , (x ∈ X → | g | x ≡ | h | x)
→
  Π a : | A | , (a ∈ Sg A X → | g | a ≡ | h | a)

```

```

hom-unique _ _ _ α a (var x) = α a x

```

```

hom-unique fe {A}{B} X g h α fa (app f a β) = | g | ((f ^ A) a) ≡ ⟨ || g || f a ⟩

```

$$\begin{aligned}
& (f \hat{=} B)(|g| \circ a) \equiv \langle \text{ap } (f \hat{=} B)(fe \text{ IH}) \rangle \\
& (f \hat{=} B)(|h| \circ a) \equiv \langle (|h| \parallel f a)^{-1} \rangle \\
& |h| ((f \hat{=} A) a) \blacksquare
\end{aligned}$$

where $\text{IH} = \lambda x \rightarrow \text{hom-unique } fe \{A\}\{B\} X g h \alpha (a x) (\beta x)$

In the induction step of the foregoing proof, the typing judgments of the premise are these:

$$\begin{array}{lll}
fe : \text{funext } \mathcal{V} \mathcal{W} & A : \text{Algebra } \mathcal{U} S & B : \text{Algebra } \mathcal{W} S \\
X : \text{Pred } |A| \mathcal{U} & g h : \text{hom } A B & \alpha : \Pi x : |A|, (x \in X \rightarrow |g| x \equiv |h| x) \\
fa : |A| & f : |S| & a : ||S|| f \rightarrow |A| \\
\beta : \text{Im } a \subseteq \text{Sg } A X
\end{array}$$

where $fa = (f \hat{=} A) a$. Under these assumptions, we prove $|g| ((f \hat{=} A) a) \equiv |h| ((f \hat{=} A) a)$.

4.2 Subalgebras

This section presents the `Subalgebras.Subalgebras` module of the `AgdaUALib`, slightly abridged.¹⁸ Here we define the `Subalgebra` type, representing the subalgebra of a given algebra, as well as the collection of all subalgebras of a given class of algebras.

4.2.1 Subalgebra type

Given algebras $A : \text{Algebra } \mathcal{U} S$ and $B : \text{Algebra } \mathcal{W} S$, we say that B is a *subalgebra* of A just in case B can be *homomorphically embedded* in A ; i.e., there exists a map $h : |B| \rightarrow |A|$ that is both a homomorphism and an embedding.¹⁹

$$\begin{aligned}
& _ \text{IsSubalgebraOf} _ : \{\mathcal{W} \mathcal{U} : \text{Universe}\} (B : \text{Algebra } \mathcal{W} S) (A : \text{Algebra } \mathcal{U} S) \rightarrow \mathcal{O} \sqcup \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W} \cdot \\
& B \text{ IsSubalgebraOf } A = \Sigma h : \text{hom } B A, \text{ is-embedding } |h| \\
& \text{Subalgebra} : \{\mathcal{W} \mathcal{U} : \text{Universe}\} \rightarrow \text{Algebra } \mathcal{U} S \rightarrow \text{ov } \mathcal{W} \sqcup \mathcal{U} \cdot \\
& \text{Subalgebra } \{\mathcal{W}\} A = \Sigma B : (\text{Algebra } \mathcal{W} S), B \text{ IsSubalgebraOf } A
\end{aligned}$$

Note the order of the arguments. The universe \mathcal{W} comes first because in certain situations we have to explicitly specify this universe, whereas we can almost always leave the universe \mathcal{U} implicit. See, for example, the definition of `IsSubalgebraOfClass` below (§4.2.3).

4.2.2 Consequences of First Homomorphism Theorem

We take this opportunity to prove an important lemma that makes use of the `IsSubalgebraOf` type defined above; it is the following: If A and B are S -algebras and $h : \text{hom } A B$ a homomorphism from A to B , then the quotient $A \text{ } \diagdown \text{ker } h$ is (isomorphic to) a subalgebra of B . This is an easy corollary of the First Homomorphism Theorem proved in the `Homomorphisms.Noether` module.

$$\begin{aligned}
& \text{FirstHomCorollary} : \{\mathcal{U} \mathcal{W} : \text{Universe}\} \\
& \quad - \text{extensionality assumptions} -
\end{aligned}$$

¹⁸For unabridged docs and source code see <https://ualib.gitlab.io/Subalgebras.Subalgebras.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Subalgebras/Subalgebras.lagda>.

¹⁹An alternative which could end up being simpler and easier to work with would be to proclaim B a subalgebra of A iff there is an *injective* homomorphism from B into A . In preparation for the next major release of the `UALib`, we will investigate the consequences of taking that path instead of the stricter embedding requirement we chose for the definition of the type `IsSubalgebraOf`.

```

→      dfunext  $\mathcal{V} \mathcal{W} \rightarrow \text{prop-ext } \mathcal{U} \mathcal{W}$ 

→      ( $\mathbf{A} : \text{Algebra } \mathcal{U} S$ )( $\mathbf{B} : \text{Algebra } \mathcal{W} S$ )( $h : \text{hom } \mathbf{A} \mathbf{B}$ )

      – truncation assumptions –

→      is-set  $|\mathbf{B}|$ 
→      ( $\forall a x \rightarrow \text{is-subsingleton } (\langle \text{kercon } \mathbf{B} h \rangle a x)$ )
→      ( $\forall C \rightarrow \text{is-subsingleton } (\mathcal{C}\{A = |\mathbf{A}|\} \{ \langle \text{kercon } \mathbf{B} h \rangle \} C)$ )
-----
→      ( $\mathbf{A} [ \mathbf{B} ] / \text{ker } h$ )  $\text{IsSubalgebraOf } \mathbf{B}$ 

```

FirstHomCorollary $fe\ pe\ \mathbf{A}\ \mathbf{B}\ h\ Bset\ ssR\ ssA = \phi_{\text{hom}}, \phi_{\text{emb}}$

where

FirstHomThm : $\Sigma \phi : \text{hom } (\mathbf{A} [\mathbf{B}] / \text{ker } h) \mathbf{B}, (| h | \equiv | \phi | \circ | \pi_{\text{ker } \mathbf{B} h} |)$
 $\times \text{Monic } | \phi | \times \text{is-embedding } | \phi |$

FirstHomThm = **FirstHomomorphismTheorem** $fe\ pe\ \mathbf{A}\ \mathbf{B}\ h\ Bset\ ssR\ ssA$

$\phi_{\text{hom}} : \text{hom } (\mathbf{A} [\mathbf{B}] / \text{ker } h) \mathbf{B}$

$\phi_{\text{hom}} = | \text{FirstHomThm} |$

$\phi_{\text{emb}} : \text{is-embedding } | \phi_{\text{hom}} |$

$\phi_{\text{emb}} = \text{snd } (\text{snd } \text{FirstHomThm})$

One special case to which we will apply this is where the algebra \mathbf{A} is the term algebra $\mathbf{T}\ X$. We formalize this special case here so that it's readily available when we need it later.

```

free-quot-subalg : { $\mathcal{U}\ \mathcal{X} : \text{Universe}$ }

      – extensionality assumptions –

→      dfunext  $\mathcal{V}\ \mathcal{U} \rightarrow \text{prop-ext } (\text{ov } \mathcal{X})\ \mathcal{U}$ 

→      ( $X : \mathcal{X} \rightarrow \cdot$ )( $\mathbf{B} : \text{Algebra } \mathcal{U} S$ )( $h : \text{hom } (\mathbf{T}\ X) \mathbf{B}$ )

      – truncation assumptions –

→      is-set  $|\mathbf{B}|$ 
→      ( $\forall p q \rightarrow \text{is-subsingleton } (\langle \text{kercon } \mathbf{B} h \rangle p q)$ )
→      ( $\forall C \rightarrow \text{is-subsingleton } (\mathcal{C}\{A = |\mathbf{T}\ X|\} \{ \langle \text{kercon } \mathbf{B} h \rangle \} C)$ )
-----
→      (( $\mathbf{T}\ X$ )  $[ \mathbf{B} ] / \text{ker } h$ )  $\text{IsSubalgebraOf } \mathbf{B}$ 

```

free-quot-subalg $fe\ pe\ X\ \mathbf{B}\ h\ Bset\ ssR\ ssB = \text{FirstHomCorollary } fe\ pe\ (\mathbf{T}\ X)\ \mathbf{B}\ h\ Bset\ ssR\ ssB$

Notation. For convenience, we define the following shorthand for the subalgebra relation.

$\underline{\leq} : \{ \mathcal{W}\ \mathcal{U} : \text{Universe} \} (\mathbf{B} : \text{Algebra } \mathcal{W}\ S) (\mathbf{A} : \text{Algebra } \mathcal{U}\ S) \rightarrow \mathbb{O} \sqcup \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W} \rightarrow \cdot$
 $\mathbf{B} \leq \mathbf{A} = \mathbf{B} \text{ IsSubalgebraOf } \mathbf{A}$

From now on we will use $\mathbf{B} \leq \mathbf{A}$ to express the assertion that \mathbf{B} is a subalgebra of \mathbf{A} .

4.2.3 Subalgebras of a class

One of our goals is to formally express and prove properties of classes of algebraic structures. Fixing a signature S and a universe \mathcal{U} , we define a class of S -algebras with domains of type $\mathcal{U} \rightarrow \cdot$ as a predicate over the $\text{Algebra } \mathcal{U}\ S$ type. In the syntax of the **UALib**, such predicates inhabit the type $\text{Pred } (\text{Algebra } \mathcal{U}\ S)\ \mathcal{X}$, for some universe \mathcal{X} .

Suppose $\mathcal{K} : \text{Pred } (\text{Algebra } \mathcal{U} \ S) \ \mathcal{E}$ denotes a class of S -algebras and $\mathbf{B} : \text{Algebra } \mathcal{W} \ S$ denotes an arbitrary S -algebra. Then we might wish to consider the assertion that \mathbf{B} is a subalgebra of some algebra in the class \mathcal{K} . The next type we define allows us to express this assertion as $\mathbf{B} \text{ IsSubalgebraOfClass } \mathcal{K}$.

$_ \text{IsSubalgebraOfClass} : \text{Algebra } \mathcal{W} \ S \rightarrow \text{Pred } (\text{Algebra } \mathcal{U} \ S) \ \mathcal{E} \rightarrow \text{ov } (\mathcal{U} \sqcup \mathcal{W}) \sqcup \mathcal{E} \ .$
 $\mathbf{B} \text{ IsSubalgebraOfClass } \mathcal{K} = \Sigma \ \mathbf{A} : \text{Algebra } \mathcal{U} \ S , \Sigma \ sa : \text{Subalgebra} \{ \mathcal{W} \} \ \mathbf{A} , (\mathbf{A} \in \mathcal{K}) \times (\mathbf{B} \cong | \ sa |)$

Using this type, we express the collection of all subalgebras of algebras in a class by the type SubalgebraOfClass , which we now define.

$\text{SubalgebraOfClass} : \{ \mathcal{W} \ \mathcal{U} : \text{Universe} \} \rightarrow \text{Pred } (\text{Algebra } \mathcal{U} \ S) (\text{ov } \mathcal{U}) \rightarrow \text{ov } (\mathcal{U} \sqcup \mathcal{W}) \ .$
 $\text{SubalgebraOfClass } \{ \mathcal{W} \} \ \mathcal{K} = \Sigma \ \mathbf{B} : \text{Algebra } \mathcal{W} \ S , \mathbf{B} \text{ IsSubalgebraOfClass } \mathcal{K}$

4.2.4 Subalgebra lemmas

We conclude this module by proving a number of useful facts about subalgebras. Some of the formal statements below may appear to be redundant, and indeed they are to some extent. However, each one differs slightly from the next, if only with respect to the explicitness or implicitness of their arguments. The aim is to make it as convenient as possible to apply the lemmas in different situations. (We're stocking the UALib utility closet now; elegance is not the priority.)

First we show that the subalgebra relation is a *preorder*. Recall, this means it is reflexive and transitive.²⁰

$\leq\text{-reflexive} : \{ \mathcal{U} : \text{Universe} \} (\mathbf{A} : \text{Algebra } \mathcal{U} \ S) \rightarrow \mathbf{A} \leq \mathbf{A}$
 $\leq\text{-reflexive } \mathbf{A} = (\text{id} \mid \mathbf{A} \mid , \text{id-is-hom} , \text{id-is-embedding})$

$\leq\text{-refl} : \{ \mathcal{U} : \text{Universe} \} \{ \mathbf{A} : \text{Algebra } \mathcal{U} \ S \} \rightarrow \mathbf{A} \leq \mathbf{A}$
 $\leq\text{-refl } \{ \mathcal{U} \} \{ \mathbf{A} \} = \leq\text{-reflexive } \mathbf{A}$

$\leq\text{-transitivity} : (\mathbf{A} : \text{Algebra } \mathcal{X} \ S) (\mathbf{B} : \text{Algebra } \mathcal{Y} \ S) (\mathbf{C} : \text{Algebra } \mathcal{Z} \ S)$
 $\rightarrow \mathbf{C} \leq \mathbf{B} \rightarrow \mathbf{B} \leq \mathbf{A} \rightarrow \mathbf{C} \leq \mathbf{A}$
 $\leq\text{-transitivity } \mathbf{A} \ \mathbf{B} \ \mathbf{C} \ CB \ BA = (\text{o-hom } \mathbf{C} \ \mathbf{A} \mid CB \mid \mid BA \mid) , \text{o-embedding} \parallel BA \parallel \parallel CB \parallel$

$\leq\text{-trans} : (\mathbf{A} : \text{Algebra } \mathcal{X} \ S) \{ \mathbf{B} : \text{Algebra } \mathcal{Y} \ S \} \{ \mathbf{C} : \text{Algebra } \mathcal{Z} \ S \}$
 $\rightarrow \mathbf{C} \leq \mathbf{B} \rightarrow \mathbf{B} \leq \mathbf{A} \rightarrow \mathbf{C} \leq \mathbf{A}$
 $\leq\text{-trans } \mathbf{A} \ \{ \mathbf{B} \} \{ \mathbf{C} \} = \leq\text{-transitivity } \mathbf{A} \ \mathbf{B} \ \mathbf{C}$

Next we prove that if two algebras are isomorphic and one of them is a subalgebra of \mathbf{A} , then so is the other.

$\leq\text{-iso} : (\mathbf{A} : \text{Algebra } \mathcal{X} \ S) \{ \mathbf{B} : \text{Algebra } \mathcal{Y} \ S \} \{ \mathbf{C} : \text{Algebra } \mathcal{Z} \ S \}$
 $\rightarrow \mathbf{C} \cong \mathbf{B} \rightarrow \mathbf{B} \leq \mathbf{A} \rightarrow \mathbf{C} \leq \mathbf{A}$
 $\leq\text{-iso } \mathbf{A} \ \{ \mathbf{B} \} \{ \mathbf{C} \} \ CB \ BA = (\mathbf{g} \circ \mathbf{f} , \mathbf{gfhom}) , \mathbf{gfemb}$
 where

²⁰ In [3], in the `Relations.Quotients` module, we defined *preorder* for binary relation types. Here, however, we will content ourselves with merely proving reflexivity and transitivity of the subalgebra relation \leq , without worry about first defining it as an inhabitant of an honest-to-goodness binary relation type, of the sort introduced in the `Relations.Discrete` module. Perhaps we will address this matter in a future release of the UALib.

```

f : | C | → | B |
f = fst | CB |
g : | B | → | A |
g = fst | BA |

gfemb : is-embedding (g ∘ f)
gfemb = o-embedding (|| BA ||) (iso→embedding CB)

gfhom : is-homomorphism C A (g ∘ f)
gfhom = o-is-hom C A {f}{g} (snd | CB |) (snd | BA |)

```

The following variations on this theme are sometimes useful.

```

≤-trans-≅ : (A : Algebra X S){B : Algebra Y S}{C : Algebra Z S}
→ A ≤ B → A ≅ C → C ≤ B
≤-trans-≅ A {B} C A ≤ B B ≅ C = ≤-iso B (≅-sym B ≅ C) A ≤ B

≤-TRANS-≅ : (A : Algebra X S){B : Algebra Y S}{C : Algebra Z S}
→ A ≤ B → B ≅ C → A ≤ C
≤-TRANS-≅ A C A ≤ B B ≅ C = ( o-hom A C | A ≤ B | | B ≅ C | ) ,
o-embedding (iso→embedding B ≅ C)(|| A ≤ B ||)

```

Next we prove a monotonicity property of \leq .

```

≤-mono : {W U X : Universe}{B : Algebra W S}{K K' : Pred (Algebra U S) X}
→ K ⊆ K' → B IsSubalgebraOfClass K → B IsSubalgebraOfClass K'
≤-mono B K K' KB = | KB | , fst || KB || , KK' (| snd || KB || ) , || (snd || KB ||) ||

```

Later we will require a number of facts having to do with the relationship between the subalgebra relation and the lifting of algebras to higher universe levels. Here are the tools we need.

```

lift-alg-is-sub : {U : Universe}{K : Pred (Algebra U S)(ov U)}{B : Algebra U S}
→ B IsSubalgebraOfClass K → (lift-alg B U) IsSubalgebraOfClass K
lift-alg-is-sub (A , (sa , (KA , B ≅ sa))) = A , sa , KA , ≅-trans (≅-sym lift-alg-≅) B ≅ sa

```

```

lift-alg-≤ : (A : Algebra X S){B : Algebra Y S} → B ≤ A → lift-alg B X ≤ A
lift-alg-≤ A {B} B ≤ A = ≤-iso A (≅-sym lift-alg-≅) B ≤ A

```

```

≤-lift-alg : (A : Algebra X S){B : Algebra Y S} → B ≤ A → B ≤ lift-alg A X
≤-lift-alg A {B} B ≤ A = ≤-TRANS-≅ B {A} (lift-alg A X) B ≤ A lift-alg-≅

```

```

lift-alg-≤-lift : {A : Algebra X S}{B : Algebra Y S} → A ≤ B → lift-alg A X ≤ lift-alg B W

```

```

lift-alg-≤-lift {A} B A ≤ B = ≤-trans (lift-alg B W) (≤-trans B IAA A ≤ B) B ≤ IB

```

where

```

IAA : (lift-alg A X) ≤ A
IAA = lift-alg-≤ A {A} ≤-refl

```

```

B ≤ IB : B ≤ lift-alg B W
B ≤ IB = ≤-lift-alg B {B} ≤-refl

```


5 Concluding Remarks

We’ve reached the end of the second installment in our three-part series describing the [AgdaUALib](#). The next part [4] covers free algebras, equational classes of algebras (i.e., varieties), and Birkhoff’s HSP theorem.

We conclude by noting that one of our goals is to make computer formalization of mathematics more accessible to mathematicians working in universal algebra and model theory. We welcome feedback from the community and are happy to field questions about the [UALib](#), how it is installed, and how it can be used to prove theorems that are not yet part of the library. Merge requests submitted to the UALib’s main gitlab repository are especially welcomed. Please visit the repository at <https://gitlab.com/ualib/ualib.gitlab.io/> and help us improve it.

References

- 1 Clifford Bergman. *Universal Algebra: fundamentals and selected topics*, volume 301 of *Pure and Applied Mathematics (Boca Raton)*. CRC Press, Boca Raton, FL, 2012.
- 2 Ana Bove and Peter Dybjer. *Dependent Types at Work*, pages 57–99. Springer Berlin Heidelberg, Berlin, Heidelberg, 2009. doi:10.1007/978-3-642-03153-3_2.
- 3 William DeMeo. The Agda Universal Algebra Library, Part 1: Foundation. *CoRR*, abs/2103.05581, 2021. Source code: <https://gitlab.com/ualib/ualib.gitlab.io>. URL: <https://arxiv.org/abs/2101.10166>, arXiv:2103.05581.
- 4 William DeMeo. The Agda Universal Algebra Library, Part 3: Identity. *CoRR*, 2021. (to appear) Source code: <https://gitlab.com/ualib/ualib.gitlab.io>. URL: http://arxiv.org/a/demeo_w_1.
- 5 Martín Hötzel Escardó. Introduction to univalent foundations of mathematics with Agda. *CoRR*, abs/1911.00580, 2019. URL: <http://arxiv.org/abs/1911.00580>, arXiv:1911.00580.
- 6 Emmanuel Gunther, Alejandro Gadea, and Miguel Pagano. Formalization of universal algebra in Agda. *Electronic Notes in Theoretical Computer Science*, 338:147 – 166, 2018. The 12th Workshop on Logical and Semantic Frameworks, with Applications (LSFA 2017). URL: <http://www.sciencedirect.com/science/article/pii/S1571066118300768>, doi:<https://doi.org/10.1016/j.entcs.2018.10.010>.
- 7 Ulf Norell. Dependently typed programming in agda. In *International school on advanced functional programming*, pages 230–266. Springer, 2008.
- 8 Ulf Norell. Dependently typed programming in Agda. In *Proceedings of the 6th International Conference on Advanced Functional Programming*, AFP’08, pages 230–266, Berlin, Heidelberg, 2009. Springer-Verlag. URL: <http://dl.acm.org/citation.cfm?id=1813347.1813352>.
- 9 The Univalent Foundations Program. *Homotopy Type Theory: Univalent Foundations of Mathematics*. Lulu and The Univalent Foundations Program, Institute for Advanced Study, 2013. URL: <https://homotopytypetheory.org/book>.
- 10 The Agda Team. Agda Language Reference section on Axiom K, 2021. URL: <https://agda.readthedocs.io/en/v2.6.1/language/without-k.html>.
- 11 The Agda Team. Agda Language Reference section on Safe Agda, 2021. URL: <https://agda.readthedocs.io/en/v2.6.1/language/safe-agda.html#safe-agda>.
- 12 The Agda Team. Agda Tools Documentation section on Pattern matching and equality, 2021. URL: <https://agda.readthedocs.io/en/v2.6.1/tools/command-line-options.html#pattern-matching-and-equality>.
- 13 Philip Wadler, Wen Kokke, and Jeremy G. Siek. *Programming Language Foundations in Agda*. July 2020. URL: <http://plfa.inf.ed.ac.uk/20.07/>.